
Primjene Coq alata za dokazivanje u matematici i
računarstvu

Hren, Miho

Master's thesis / Diplomski rad

2024

Degree Grantor / Ustanova koja je dodijelila akademski / stručni stupanj: University of 
Zagreb, Faculty of Electrical Engineering and Computing / Sveučilište u Zagrebu, Fakultet 
elektrotehnike i računarstva

Permanent link / Trajna poveznica: https://urn.nsk.hr/urn:nbn:hr:168:192892

Rights / Prava: In copyright / Zaštićeno autorskim pravom.

Download date / Datum preuzimanja: 2025-03-14

Repository / Repozitorij:

FER Repository - University of Zagreb Faculty of 
Electrical Engineering and Computing repozitory

https://urn.nsk.hr/urn:nbn:hr:168:192892
http://rightsstatements.org/vocab/InC/1.0/
http://rightsstatements.org/vocab/InC/1.0/
https://repozitorij.fer.unizg.hr
https://repozitorij.fer.unizg.hr
https://zir.nsk.hr/islandora/object/fer:12332
https://repozitorij.unizg.hr/islandora/object/fer:12332
https://dabar.srce.hr/islandora/object/fer:12332


SVEUČILIŠTE U ZAGREBU

FAKULTET ELEKTROTEHNIKE I RAČUNARSTVA

DIPLOMSKI RAD br. 578

PRIMJENE COQ ALATA ZA DOKAZIVANJE U MATEMATICI I

RAČUNARSTVU

Miho Hren

Zagreb, lipanj 2024.

 



SVEUČILIŠTE U ZAGREBU

FAKULTET ELEKTROTEHNIKE I RAČUNARSTVA

DIPLOMSKI RAD br. 578

PRIMJENE COQ ALATA ZA DOKAZIVANJE U MATEMATICI I

RAČUNARSTVU

Miho Hren

Zagreb, lipanj 2024.

 



SVEUČILIŠTE U ZAGREBU
FAKULTET ELEKTROTEHNIKE I RAČUNARSTVA

Zagreb, 4. ožujka 2024.

DIPLOMSKI ZADATAK br. 578

Pristupnik: Miho Hren (0036513445)

Studij: Računarstvo

Profil: Računarska znanost

Mentor: izv. prof. dr. sc. Ante Đerek

Zadatak: Primjene Coq alata za dokazivanje u matematici i računarstvu

Opis zadatka:

Coq je alat za interaktivno razvijanje formalnih dokaza u matematici i računarstvu zasnovan na takozvanom
računu induktivnih konstrukcija. U sklopu diplomskog rada potrebno je, koristeći alat Coq, implementirati logiku
prvog reda s induktivnim definicijama FOL_ID i cikličkim dokazima u tom kontekstu, oslanjajući se na
Brothersonov članak "Cyclic Proofs for First-Order Logic with Inductive Definitions". Dalje je potrebno pokušati
formalizirati neke nerazrađene detalje u mehaničkoj certifikaciji cikličkih dokaza u FOL_ID. Ako je moguće,
potrebno je i istražiti kako se ciklički dokazi mogu upotrijebiti za automatsko dokazivanje terminacije programa.
Ako je moguće, potrebno je formalizirati teoriju Rowea i Brothersona iz članka "Automatic cyclic termination
proofs for recursive procedures in separation logic" u Coqu te krenuti prema implementaciji verificiranog alata za
generiranje dokaza terminacije. Radu je potrebno priložiti izvorni kod razvijenih i korištenih programa, citirati
korištenu literaturu i navesti dobivenu pomoć.

Rok za predaju rada: 28. lipnja 2024.



Wir müssen wissen — wir werden wissen!
Mentori Vedran Čačić i Marko Doko omogućili su mi proučavanje najzanimljivijih tema.

Hvala Vam!



Sadržaj

Uvod . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3

1. Coq . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4

1.1. Što je Coq? . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4

1.2. Programiranje u Coqu . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6

1.3. Hijerarhija tipova . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9

1.4. Propozicije i tipovi, dokazi i termi . . . . . . . . . . . . . . . . . . . . . . 10

1.5. Ograničenja tipskog sustava . . . . . . . . . . . . . . . . . . . . . . . . . . 12

2. Logika prvog reda s induktivnim definicijama . . . . . . . . . . . . . . . 15

2.1. Sintaksa . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 16

2.2. Semantika . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 23

2.3. Standardni modeli . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 26

3. Sistem sekvenata s induktivnim definicijama . . . . . . . . . . . . . . . 31

3.1. Strukturalna pravila . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33

3.2. Propozicijska pravila . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34

3.3. Pravila za kvantifikatore . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35

3.4. Pravila indukcije . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 36

3.5. Produkcijska pravila . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 39

3.6. Dokaz i teorem . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 39

3.7. Adekvatnost . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 41

4. Ciklički dokazi . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 43

4.1. Primjer: lijene liste . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 43

4.2. Primjer: 𝐶𝐿𝐾𝐼𝐷𝜔 . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 45

1



Zaključak . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 48

Literatura . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 49

Sažetak . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 51

Abstract . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 52

2



Uvod

Coq je interaktivni dokazivač teorema u kojem korisnik iskazujematematičke tvrdnje ti-

povima te ih dokazuje programiranjem. Teorija tipova koju Coq implementira temelji se

na računu induktivnih konstrukcija koji pak je proširenje 𝜆-računa tipovima. Iskaziva-

nje tvrdnji moguće je zahvaljujući sorti Prop te zavisnim tipovima. Zahvaljujući jeziku

taktika moguće je i automatizirano dokazivanje nekih tvrdnji, a osim za dokazivanje,

Coq se može koristiti i za pisanje dokazano točnih programa.

Središnji pojam ovog rada je logika prvog reda s induktivnim definicijama 𝐹𝑂𝐿𝐼𝐷,

koju je prvi uveo James Brotherston, a u kojoj se neki predikati definiraju na induk-

tivni način produkcijama. Zbog induktivnih predikata nužno je sa semantičkog stajališta

promatrati posebne vrste struktura u kojima interpretacije induktivnih predikata imaju

smisla s obzirom na njihove definicije. Sa sintaksnog stajališta promatra se Gentzenov

sistem sekvenata proširen pravilima za induktivne predikate.

Ovaj rad sastoji se od četiri dijela. U prvom dijelu prikazujemo osnovne Coqa kao

programskog jezika te osnove njegove teorijske pozadine, prvenstveno hijerarhije tipova

i Curry–Howardove korespondencije. U drugom dijelu čitatelj se upoznaje s logikom

𝐹𝑂𝐿𝐼𝐷 kroz pojmove formula, struktura te standardnih modela. U trećem dijelu de-

finiramo dokazni sustav 𝐿𝐾𝐼𝐷 za logiku 𝐹𝑂𝐿𝐼𝐷 te dokazujemo neka njegova svojstva.

Sve definicije i tvrdnje u drugom i trećem dijelu popraćene su svojim formalizacijama

u Coqu. Konačno, u četvrtom dijelu ilustriramo pojam cikličkog dokaza primjerima u

Coqu i u dokaznom sustavu 𝐶𝐿𝐾𝐼𝐷𝜔. Cijeli rad prožet je ilustrativnim primjerima, kako

neformalnih definicija i iskaza, tako i njihovih formalnih reprezentacija u Coqu.

1. Coq

U ovom poglavlju dajemo pregled visoke razine programskog sustava Coq. Prvo ćemo

objasniti što je uopće Coq, u kojem je kontekstu nastao, i od kojih komponenti se sas-

toji. Zatim ćemo dati kratak pregled programiranja u Coqu, nakon čega ćemo se baviti

naprednijim konceptima i spomenuti neka ograničenja. Za širi opseg gradiva, čitatelja

upućujemo na knjige Coq’Art [1], Software Foundations [2, 3, 4] i Certified Programming

with Dependent Types [5] te na službenu dokumentaciju [6].

1.1. Što je Coq?

Alat za dokazivanje Coq1, punog naziva The Coq Proof Assistant, programski je sustav

pomoću kojeg korisnici mogu dokazivati matematičke tvrdnje, a može se koristiti i kao

funkcijski programski jezik sa zavisnim tipovima. Alat se temelji na 𝜆-računu i teoriji

tipova, a prva je inačica implementirana 1984. godine [6]. Ovaj rad koristi inačicu 8.18

iz rujna 2023. godine.

ProgramCoqmože se pokrenuti u interaktivnom ili u skupnomnačinu rada. Interak-

tivni način rada pokreće se naredbom coqtop, a korisniku omogućuje rad u ljusci sličnoj

ljuskama bash i python. Interaktivna ljuska (također poznata pod imenom toplevel) služi

unosu definicija te iskazivanju i dokazivanju tvrdnji. Skupni način rada pokreće se na-

redbom coqc, a korisniku omogućuje provjeru tipova i prevođenje izvornih datoteka u

strojno čitljive formate. Kod formaliziranja i dokazivanja, korisnik će najčešće koristiti

interaktivni način rada, po mogućnosti kroz neku od dostupnih razvojnih okolina.2

Kao programski jezik, Coq se sastoji od više podjezika različitih namjena, od kojih

1https://coq.inria.fr/
2Autor rada koristio je paket Proof General za uređivač teksta Emacs. Druge često korištene okoline su

VsCoq i CoqIDE.

spominjemo Vernacular, Gallinu i Ltac.

Vernacular („govorni jezik”) je jezik naredbi kojima korisnik komunicira sa sustavom

(i u interaktivnom i u skupnom načinu rada); svaka Coqova skripta (datoteka s

nastavkom .v) je niz naredbi. Često korištene naredbe su Check, Definition,

Inductive, Fixpoint i Lemma. Pomoću naredbi za iskazivanje tvrdnji, kao što je

Lemma, Coq ulazi u način dokazivanja (proof mode).

Gallina je Coqov strogo statički tipiziran specifikacijski jezik. Dokazi svih tvrdnji pred-

stavljeni su interno kao programi u Gallini. Kako se glavnina programiranja u

Coqu svodi upravo na programiranje u Gallini, posvećujemo joj idući odjeljak.

Ltac je Coqov netipizirani jezik za definiciju i korištenje taktika. Taktike su pomoćne

naredbe kojima se u načinu dokazivanja konstruira dokaz. Može se reći da je Ltac

jezik za metaprogramiranje Galline. Primjeri taktika su intros, destruct, apply

i rewrite.

Pogledajmo primjer za ilustraciju odnosa ta tri jezika.

1 Lemma example_lemma: 1 + 1 = 2.

2 Proof.

3 cbn. reflexivity.

4 Qed.

Ključne riječi Lemma, Proof i Qed dio su Vernaculara, izraz example_lemma: 1+1=2 dio

je Galline, a pomoćne naredbe cbn i reflexivity dio su Ltaca.

Jezgra je programskog sustava Coq algoritam za provjeru tipova (type checking) —

svaka tvrdnja koja se dokazuje iskazana je tipovima. Ostatak sustava u načelu služi za

knjigovodstvo i poboljšanje korisničkog iskustva.3 Nužno je da jezgra sustava bude rela-

tivno mala kako bismo se mogli uvjeriti u njenu točnost. U suprotnom, možemo li biti

sigurni da su naše dokazane tvrdnje doista istinite?

Prve inačice Coqa implementirale su samo račun konstrukcija [7] koji je proširenje

𝜆-računa polimorfnim i zavisnim tipovima te tipskim konstruktorima. Kasnije je dodana

podrška za induktivno i koinduktivno definirane tipove [8, 9], a danas se može reći da

Coq implementira polimorfni kumulativni račun induktivnih konstrukcija [10]. Coq se,

osim kao dokazivač teorema, može koristiti i za programiranje sa zavisnim tipovima. U

3I jezgra i ostatak sustava implementirani su u OCamlu.

5



toj sferi još su istaknuti jezici Agda4, Idris5 i Lean6. Coq se između njih ističe po usmje-

renosti prema dokazivanju, posebno po korištenju taktika (jezik Ltac) i nepredikativnoj

sorti Prop (o kojoj će kasnije biti riječi). Još jedna prednost Coqa je mehanizam ekstrak-

cije pomoću kojeg korisnik može proizvoljnu funkciju prevesti u jezik niže razine aps-

trakcije.7 Mehanizam ekstrakcije nije dokazano točan, no poželjno je da izvorne funkcije

budu ekvivalentne ekstrahiranima pa se radi na verifikaciji ekstrakcije [10].

1.2. Programiranje u Coqu

Gallina je funkcijski programski jezik, što znači da su funkcije prvoklasni objekti — one

mogu biti argumenti i povratne vrijednosti drugih funkcija. Dodatno, iteracija se ostva-

ruje rekurzijom te ne postoje tradicionalne varijable, već se koriste nepromjenjiva (im-

mutable) imena. Za uvod u funkcijsko programiranje, čitatelja upućujemona knjigu Pro-

gramming in Haskell [11]. Primjeri koje ćemo vidjeti u ostatku ovog odjeljka oslanjanju

se na tipove i funkcije definirane u Coqovoj standardnoj knjižnici.8

Gallina je strogo statički tipiziran jezik, što znači da se svakom termu prilikom prevo-

đenja dodjeljuje tip9. Naredbom Checkmožemo provjeriti tip nekog terma ili doznati da

termu Coq ne može dodijeliti tip. Dalje u radu pod „term” mislimo na dobro formirane

terme, odnosno na one kojima Coqmože dodijeliti tip. Kažemo da je term stanovnik tipa

koji mu je dodijeljen. Za tip kažemo da je nastanjen, odnosno nenastanjen, ako postoji,

odnosno ne postoji, stanovnik tog tipa. Kako su u Coqu i tipovi termi, radi razumljivosti

i zvučnosti umjesto „tip tipa” kažemo „sorta tipa”.

Kao i u ostalim jezicima, kod programiranja u Coqu korisnik se oslanja na dostupne

primitivne izraze, od kojih su najvažniji:

• forall za konstrukciju funkcijskih tipova i zavisnih produkata,

• match za rad sa stanovnicima induktivnih tipova te

• fun, fix i cofix za definiciju funkcija.

Naredbom Inductive definira se induktivni tip te se automatski za njega generiraju

4https://wiki.portal.chalmers.se/agda/
5https://www.idris-lang.org/
6https://lean-lang.org/
7Trenutno su podržani Haskell, OCaml i Scheme.
8https://coq.inria.fr/library/
9Tipovi su kolekcije objekata na kojima je moguće provoditi srodne operacije.

principi indukcije i rekurzije.

1 Inductive nat : Set :=

2 | O : nat

3 | S : nat -> nat.

Ovim kodom definirali smo tri terma:

• nat (tip prirodnih brojeva) je term sorte Set,

• O (broj nula) je term tipa nat i

• S (funkcija sljedbenika) je term tipa nat→nat.

Za term nat kažemo da je konstruktor tipa (type constructor), a za terme O i S kažemo da

su konstruktori objekata (object constructors).

Jedna od osnovnih naredbi za imenovanje novih terma je naredba Definition.

1 Definition negb (b : bool) : bool :=

2 match b with

3 | false => true

4 | true => false

5 end.

U gornjem kodu definirana je funkcija negb čiji se argument b tipa bool destrukturira

te se vraća njegova negacija, također tipa bool. Važno je napomenuti da izrazi koji po-

činju s match t ((pattern matching)), gdje je t stanovnik tipa T, moraju imati po jednu

granu za svaki konstruktor tipa T.10 U ovom su primjeru konstante false i true jedini

konstruktori tipa bool. Funkcija negb je tipa bool→bool.

1 Definition mult_zero_r : Prop := forall (n : nat), n * 0 = 0.

Ovdje je definirana propozicija (tip) imena mult_zero_r kao tvrdnja univerzalno kvan-

tificirana po prirodnim brojevima.

Rekurzija nad induktivnim tipovima može se ostvariti naredbom Fixpoint, koja u

pozadini koristi naredbu Definition te izraz fix.

10U sprezi s uvjetom strukturalne rekurzije, ovime je osigurana totalnost svake funkcije.

1 Fixpoint plus (n m : nat) {struct n} : nat :=

2 (* Definition plus := fix plus (n m : nat) {struct n} : nat := *)

3 match n with

4 | O => m

5 | S n' => S (plus n' m)

6 end.

U ovom primjeru definirana je funkcija plus koja prima dva argumenta tipa nat.

Funkcija je rekurzivna s obzirom na prvi argument što je vidljivo iz oznake {struct n}.

Napominjemo da su induktivni tipovi dobro utemeljeni, to jest svaki term induktivnog

tipa je konačan.

Osim induktivnih, u Coqu postoje i koinduktivni tipovi, koji nisu dobro utemeljeni,

zbog čega za njih nije moguće definirati principe indukcije i rekurzije. Umjesto rekur-

zije, koinduktivni tipovi koriste se u korekurzivnim funkcijama. Standardni primjer ko-

induktivnog tipa je beskonačna lista.

1 Set Primitive Projections.

2 CoInductive Stream (A : Type) :=

3 Cons { hd : A; tl : Stream A; }.

Ovime smo definirali familiju tipova Stream indeksiranu tipskom varijablom A. Svaki

Stream ima glavu i rep koji je također Stream.

Stanovnici koinduktivnih tipova konstruiraju se korekurzivnim funkcijama nared-

bom CoFixpoint, koja u pozadini koristi naredbu Definition te izraz cofix.

1 CoFixpoint from (n : nat) : Stream nat := Cons _ n (from (S n)).

2 (* Definition from := cofix from (n : nat) := Cons _ n (from (S n)) *)

Ovime je definirana funkcija from koja za ulazni argument n vraća niz prirodnih brojeva

od 𝑛 na dalje.

Razlika induktivnih i koinduktivnih tipova može se sumirati epigramom:

„Induktivni tipovi su domene rekurzivnih funkcija,

koinduktivni tipovi su kodomene korekurzivnih funkcija”.

Time se želi reći da se termi induktivnih tipova destrukturiraju u rekurzivnim funkci-

jama, dok se termi koinduktivnih tipova konstruiraju u korekurzivnim funkcijama.

1.3. Hijerarhija tipova

U usporedbi s tradicionalnim programskim jezicima, Coqov tipski sustav je ekspresiv-

niji jer dopušta tipove koji mogu ovisiti o termima. Takvi tipovi se u Coqu konstruiraju

izrazom oblika forall. Primjer jednog takvog tipa je „lista duljine 𝑛”, gdje je 𝑛 neki pri-

rodan broj. Njegovi su stanovnici 𝑛-torke, a on ovisi o stanovniku drugog tipa (u ovom

slučaju, o n tipa nat).

Spomenuli smo da su i tipovi termi te im se može dodijeliti sorta. Postoji li najveća

sorta, odnosno postoji li tip Type čiji su stanovnici svi tipovi? Prisjetimo se, svaki term

ima svoj tip. Kada bi takav Type postojao, tada bi vrijedilo Type:Type, što može dovesti

do paradoksa samoreferenciranja.11 Takav dokazivač teorema bio bi inkonzistentan te

bismo njime mogli dokazati kontradikciju, čime dokazivač efektivno gubi svoju svrhu.

Umjesto jedne „velike” sorte Type, u Coqu postoji rastući niz sorti Type𝑛 za sve prirodne

brojeve 𝑛, takav da vrijedi Type𝑛 ∶ Type𝑚 kad god vrijedi 𝑛 < 𝑚. Ilustracija ove kumu-

lativne hijerarhije tipova prikazana je na slici 1.1. Dvije najvažnije sorte u Coqu su

Set i Prop.

Naziv Set sinonim je za sortu Type0, a njene stanovnike zovemomalim tipovima.

Primjerice, tipovi nat i bool su mali tipovi. Dodatno, tipovi funkcija koje primaju i vra-

ćaju male tipove su također mali tipovi. Također su i produkti, sume, liste i stabla malih

tipova ponovomali tipovi. Intuitivno se može reći da su mali oni tipovi s čijim se stanov-

nicima može efektivno računati. Stanovnike malih tipova nazivamo programima.

Stanovnici sorte Prop su propozicije (izjave). Za razliku od programa, s propozici-

jama ne možemo efektivno računati, ali ih možemo dokazivati. Stanovnici propozicija

su njihovi dokazi.

Za dokazivače teorema, poželjna je mogućnost definicije predikata (propozicija) nad

proizvoljnim tipovima. Zbog toga u Coqu prilikom definicije terma sorte Propmožemo

kvantificirati po proizvoljno velikim tipovima, što uključuje i sortu Prop. Tako je ovdje

not predikat nad sortom Prop.

1 Definition not (P : Prop) := forall (Q : Prop), P -> Q.

11U naivnoj logici to je Epimenidov paradoks („Ova je rečenica lažna.”), a u naivnoj teoriji skupova to
je Russellov paradoks, u naivnoj teoriji tipova to je Girardov paradoks.

⋮

Type𝑚+1

. . . bool→Type𝑚 Type𝑚

. . . ⋮

Type1

Set

bool

nat

⋮

list nat

prod nat bool

nat→nat

⋮

. . .

bool→Set . . . Prop

True

1+1=2

⋮

False

∀b,negb b=b

⋮

. . .

nat→Prop . . .

. . .

Type𝑚 → Prop . . .

Slika 1.1. Kumulativna hijerarhija tipova

Valja primijetiti da kod definicije predikata not kvantificiramo po Prop, no za svaku

propoziciju P je term not P također tipa Prop, pa domena funkcije not P uključuje

sam term not P. Ovaj stil kvantifikacije omogućuje nam definiciju proizvoljnih propo-

zicijskih veznika.12 Kažemo da je sorta Prop nepredikativna. S druge strane, sorta

Set je predikativna, to jest ne dopušta kvantifikaciju po Set i ostalim proizvoljno veli-

kim tipovima. Kada bi sorta Prop bila predikativna, ne bismo mogli definirati predikat

not: Prop → Prop.

1.4. Propozicije i tipovi, dokazi i termi

U decimalnom zapisu broja 𝜋, barem jedna znamenka pojavljuje se beskonačno mnogo

puta. Doista, kada bi se svaka znamenka pojavljivala samo konačno mnogo puta, broj

𝜋 bio bi racionalan. Međutim, nije jasno koja znamenka ima to svojstvo. Možda ih ima

više. Štoviše, vjerujemo da su sve znamenke takve. Da bismo odgovorili na to pitanje,

12U praksi su veznici definirani induktivno.

morali bismo prebrojiti sve znamenke broja𝜋, što nijemoguće u konačnomnogo koraka.

Sličnim pitanjima bavili su se logičari dvadesetog stoljeća. Klasični logičari bi gornju

tvrdnju smjesta prihvatili, dok bi konstruktivisti tražili konkretnu znamenku. Između

ostalog, ovakva razmatranja rezultirala su fundamentalnim uvidom u povezanost pro-

gramiranja i dokazivanja. Naime, želimo li dokazati konjunkciju, dovoljno je zasebno

dokazati njene konjunkte. S druge strane, želimo li konstruirati par objekata, dovoljno

je zapakirati prvi i drugi objekt u konstruktor para. Na sličan način, želimo li dokazati

implikaciju, dovoljno je pretpostaviti njen antecedent te pomoću njega dokazati kon-

zekvens. Ako pak želimo konstruirati funkciju, smijemo uzeti njen argument i pomoću

njega konstruirati povratnu vrijednost. Dodatno, nemoguće je dokazati laž, a istina trivi-

jalno vrijedi. S druge strane, ako induktivni tip nema konstruktore, onda je nenastanjen

jer nije moguće definirati vrijednost tog tipa. Ako pak tip ima barem jedan konstruktor

bez argumenata, tada postoje i njegovi stanovnici. Kroz ove primjere vidimo fenomen

Curry–Howardove korespondencije, koju možemo sažeti epigramom:

„Propozicije su tipovi, dokazi su programi.”

Time se dokazivanje svodi na programiranje. Pogledi na dvije strane ovog „novčića”

mogu se vidjeti u tablici 1.1.

Dokazivanje Programiranje

propozicija tip
dokaz program

laž prazan tip
istina nastanjen tip

konjunkcija produktni tip
disjunkcija zbrojni tip
implikacija funkcijski tip

univerzalna kvantifikacija zavisni produkt
egzistencijalna kvantifikacija zavisna suma

Tablica 1.1. Sličnosti dokazivanja i programiranja

Za bolju ilustraciju, prikazujemo princip matematičke indukcije u Coqu. Prisjetimo

se, za proizvoljni predikat 𝑃 na prirodnim brojevima, princip matematičke indukcije

glasi:

𝑃(0) ∧ ∀𝑛, (𝑃(𝑛)→ 𝑃(𝑛 + 1))→ ∀𝑛, 𝑃(𝑛).

11



Tvrdnju dokazujemo analizom broja 𝑛. Ako je 𝑛 = 0, tvrdnja slijedi iz baze indukcije.

Ako je pak 𝑛 = 𝑛′+1 za neki 𝑛′, tada rekurzivno konstruiramo dokaz za 𝑃(𝑛′), a tražena

tvrdnja slijedi primjenom koraka indukcije na rekurzivno konstruirani dokaz.

1 Definition nat_ind (P : nat -> Prop)

2 (baza : P 0)

3 (korak : forall m, P m -> P (S m))

4 : forall n, P n :=

5 fix F (n : nat) : P n :=

6 match n with

7 | O => baza

8 | S m => korak m (F m)

9 end.

Za term nat_ind kažemo da je dokazni term (proof term) za tvrdnju matematičke

indukcije. Princip matematičke indukcije je samo poseban slučaj principa indukcije,

koji Coq automatski generira za svaki induktivno definiran tip pri njegovoj definiciji.

1.5. Ograničenja tipskog sustava

Kao što smo već vidjeli, Coqov tipski sustav je izražajniji od tipskih sustava uobičajenih

programskih jezika. Međutim, kako bi se sačuvala poželjna svojstva algoritma provjere

tipova, ipak se tipski sustav mora ograničiti.

Uvjet pozitivnosti odnosi se na definiciju induktivnih i koinduktivnih tipova. Ovo

ograničenje zabranjuje negativne pojave tipa kojeg definiramo u argumentima njegovih

konstruktora.

1 Inductive Lam :=

2 | LamVar (n : nat)

3 | LamApp (M N : Lam)

4 | LamAbs (M : Lam -> Lam).

Pokretanje primjera iznad rezultira greškom Non strictly positive occurrence of

"Lam" in "(Lam -> Lam) -> Lam" — drugim riječima, tip Lam se javlja negativno u

konstruktoru LamAbs, odnosno kao argument funkcije koja je parametar konstruktora.

Ovaj uvjet štiti korisnika od inkonzistentnosti, a za točnu definiciju pozitivnosti čitate-

lja upućujemo na dokumentaciju.13 Uz uvjet pozitivnosti za induktivne tipove vezan je

13https://coq.inria.fr/doc/v8.18/refman/language/core/inductive.html#well-formed-

inductive-definitions

uvjet strukturalne rekurzije. Ovim uvjetom osigurava se totalnost rekurzivno defini-

rane funkcije tako da se argument po kojem je funkcija rekurzivna strukturalno smanjuje

u svakomkoraku rekurzije (funkcija se poziva samo na pravompodtermu originalnog ar-

gumenta).

Uvjet produktivnosti odnosi se na definiciju korekurzivnih funkcija, a dualan je

uvjetu strukturalne rekurzije. Ovaj uvjet također štiti korisnika od inkonzistentnosti, a

glasi: svaki korekurzivni poziv smije se pojaviti samo kao izravni argument konstruktora

koinduktivnog tipa čiji element definiramo (poziv funkcije mora stvoriti pravi nadterm

originalnog poziva).14 Zbog tog uvjeta, iduća definicija nije moguća.

1 Set Primitive Projections.

2 CoInductive NatStream := {

3 nat_hd : nat;

4 nat_tl : NatStream;

5 }.

6

7 CoFixpoint foo : NatStream := foo.

Greška koju sustav javlja glasi Unguarded recursive call in "foo", što znači da

se korekurzivni poziv foo ne javlja kao izravni argument konstruktora. S druge strane,

definicija

1 CoFixpoint bar : NatStream := {| nat_hd := 0; nat_tl := bar |}.

je sasvim legalna.

Posljednje ograničenje koje spominjemo vezano je uz irelevantnost dokaza (proof ir-

relevance). Naime, mnogi teoremi mogu se dokazati na više načina, ali pojedini dokaz

(dakle, postupak kojim smo od pretpostavki došli do konkluzije) nije bitan. Matemati-

čarima su bitni samo iskaz teorema i činjenica da se teorem može dokazati. Sam postu-

pak dokazivanja smatra se „implementacijskim detaljem”. Upravo zato analiza dokaza

ima smisla samo kada se dokazuje, ali ne i kada se programira. U našoj terminologiji to

znači da se pattern matching nad dokazima smije provoditi samo kod definiranja terma

sorte Prop. U suprotnom, mogli bismo definirati programe koji ovise o konkretnom do-

kazu, umjesto o iskazanom teoremu. Ograničenje eliminacije propozicije nastalo je

14https://coq.inria.fr/doc/v8.18/refman/language/core/coinductive.html#co-

recursive-functions-cofix

radi omogućavanja ekstrakcije — svi termi sorte Prop se „brišu” prilikom prevođenja iz

Coqovog tipskog sustava u tipske sustave niže razine apstrakcije. Kada ovog ograničenja

ne bi bilo, ekstrakcija u jednostavnije jezike naprosto ne bi bila moguća, jer bi prilikom

ekstrakcije bilo nužno zadržati i sve dokaze uz svu kompleksnost njihovih tipova.

2. Logika prvog reda s induktivnim

definicijama

U ovom poglavlju predstavljamo glavne rezultate diplomskog rada: formalizaciju logike

prvog reda s induktivnim definicijama 𝐹𝑂𝐿𝐼𝐷 te dokaznog sustava 𝐿𝐾𝐼𝐷, koje je prvi

uveo Brotherston [12]. Definicije, leme i dokazi u ovom poglavlju preuzete su iz Brot-

herstonove disertacije [13]. Za općeniti uvod u logiku čitatelja upućujemo na knjigu

Matematička logika [14].

Prvo ćemo definirati sintaksu i semantiku logike 𝐹𝑂𝐿𝐼𝐷, nakon čega ćemo definirati

njene standardne modele. Zatim ćemo prikazati dokazni sustav 𝐿𝐾𝐼𝐷 te konačno doka-

zati dio adekvatnosti sustava 𝐿𝐾𝐼𝐷 s obzirom na standardnu semantiku, što je ujedno i

glavni rezultat ovog diplomskog rada.

Svaka definicija i lema u ovom poglavlju bit će popraćena svojom formalizacijom u

Coqu. Jedan je od ciljeva diplomskog rada prikazati primjene Coqa u matematici, zbog

čega leme nećemo dokazivati „na papiru”, već se dokaz svake lememože pronaći u repo-

zitoriju rada.1 Zainteresiranom čitatelju predlažemo interaktivni prolazak kroz dokaze

lema u nekoj od razvojnih okolina za Coq.

Prije no što krenemo na formalizaciju, valja prokomentirati odnos matematičkog i

Coqovog vokabulara što se tiče riječi „skup”. U matematici pojam „skup” može imati

dva značenja; prvo se odnosi na skupove kao domene diskursa, dok se drugo odnosi na

skupove kao predikate, odnosno podskupove domene diskursa. Primjerice, skup prirod-

nih brojeva ℕ je domena diskursa kada je riječ o svim prirodnim brojevima te zbog toga

pišemo 𝑛 ∈ ℕ umjesto ℕ(𝑛). S druge strane, skup svih parnih prirodnih brojeva 𝐸 je

podskup skupaℕ, a može se interpretirati kao predikat parnosti na prirodnim brojevima

1https://github.com/mihohren/diploma-thesis

te možemo pisati 𝐸(𝑛) umjesto 𝑛 ∈ 𝐸. U Coqu se skupovi kao domene diskursa for-

maliziraju tipovima sorte Set2, dok se skupovi kao predikati formaliziraju funkcijama

iz domene diskursa u sortu Prop. Na primjer, tip prirodnih brojeva nat je sorte Set, a

predikat Nat.Even je tipa nat→Prop.

2.1. Sintaksa

Kao i u svakom izlaganju logike, na početku je potrebno definirati sintaksu.

Definicija 1. Signatura prvog reda s induktivnim predikatima (kratko: signatura), s oz-

nakom Σ, je skup simbola od kojih razlikujemo funkcijske, obične predikatne i induktivne

predikatne simbole. Mjesnost simbola reprezentiramo funkcijom iz odgovarajućeg skupa

simbola u skup ℕ, a označujemo ju s |𝑓| za funkcijske, odnosno s |𝑃| za predikatne sim-
bole.

1 Structure signature := {

2 FuncS : Set;

3 fun_ar : FuncS -> nat;

4 PredS : Set;

5 pred_ar : PredS -> nat;

6 IndPredS : Set;

7 indpred_ar : IndPredS -> nat;

8 }.

Primjer 1. Funkcijski simboli Peanove aritmetike su redom simbol nule 𝑜 (mjesnosti 0),

simbol sljedbenika 𝑠 (mjesnosti 1) te simboli zbroja + i umnoška ⋅ (mjesnosti 2).

1 Inductive Func__PA :=

2 | PA_zero

3 | PA_succ

4 | PA_add

5 | PA_mult.

6 Definition fun_ar__PA (s : Func__PA) : nat :=

7 match s with

8 | PA_zero => 0

9 | PA_succ => 1

10 | PA_add => 2

11 | PA_mult => 2

12 end.

2Ili tipovima sorte Type kada nam trebaju veliki tipovi, najčešće kao posljedica parametrizacije opće-
nitih tipova.

Simbol jednakosti= (mjesnosti 2) je jedini obični predikatni simbol Peanove aritmetike.

1 Inductive Pred__PA := PA_eq.

2 Definition pred_ar__PA (s : Pred__PA) : nat := 2.

Definiramo induktivne predikatne simbole 𝑁𝑎𝑡, 𝐸𝑣𝑒𝑛 i 𝑂𝑑𝑑 mjesnosti 1 koje ćemo in-

terpretirati redom kao predikate „biti prirodan”, „biti paran” te „biti neparan”.

1 Inductive IndPred__PA :=

2 | PA_Nat

3 | PA_Even

4 | PA_Odd.

5 Definition indpred_ar__PA (s : IndPred__PA) : nat := 1.

Konačno, proširena Peanova signatura, s oznakom Σ𝑃𝐴, je signatura za Peanovu aritme-

tiku prvog reda s prethodno definiranim induktivnim predikatnim simbolima.

1 Definition Σ__PA : signature

2 := {|

3 FuncS := Func__PA;

4 fun_ar := fun_ar__PA;

5 PredS := Pred__PA;

6 pred_ar := pred_ar__PA;

7 IndPredS := IndPred__PA;

8 indpred_ar := indpred_ar__PA;

9 |}.

U ostatku poglavlja promatramo jednu proizvoljnu, ali fiksiranu signaturu Σ. Fik-

siranje nekog proizvoljnog objekta je česta pojava u matematici, prvenstveno zato što

fiksirane argumente ne trebamo spominjati eksplicitno u kasnijim definicijama i iska-

zima. Coq omogućuje fiksiranje naredbom Context, pod uvjetom da se korisnik nalazi

u okolini Section.3 Većina definicija i lema u ovom radu su napisane upravo unutar

takvih okolina.

Definicija 2. Varijabla je prirodan broj. Skup svih terma konstruiramo rekurzivno na

način:

1. svaka varijabla je term;

2. ako je 𝑓 funkcijski simbol mjesnosti 𝑛 te su 𝑡1, . . . , 𝑡𝑛 termi
4, onda je 𝑓(𝑡1, . . . , 𝑡𝑛)

također term.
3https://coq.inria.fr/doc/v8.18/refman/language/core/sections.html
4Primijetimo, broj terma ovisi o mjesnosti funkcijskog simbola. U Coqovoj implementaciji ovog „kons-

truktora” možemo vidjeti da je on zavisnog tipa.

1 Inductive term : Set :=

2 | var_term : var -> term

3 | TFunc : forall (f : FuncS Σ), vec term (fun_ar f) -> term.

Uobičajene prezentacije logike prvog reda za skup varijabli uzimaju proizvoljni skup 𝒱 ,

no za formalizaciju je pogodniji skup prirodnih brojeva ℕ. Umjesto eksplicitne kvan-

tifikacije po nekoj varijabli 𝑣, implicitno ćemo kvantificirati po varijabli 0. Ovaj pristup

kvantifikaciji5, nazvan „de Bruijnovo indeksiranje” [15], bitno olakšava rad sa supstituci-

jama. O samoj implementaciji de Bruijnovog indeksiranja više se može pročitati u knjizi

Types and Programming Languages [16]. Za potrebe ovog rada koristili smo program

Autosubst26 [17, 18] za automatsko generiranje tipova terma i formula te pripadajućih

funkcija supstitucija i pomoćnih lema.

Princip indukcije za term potrebno je ručno definirati. Naime, induktivni tip term je

ugniježđen po konstruktoru TFunc zato što se javlja omotan oko drugog induktivnog tipa7

kao argument. Za ugniježđene induktivne tipove, Coq generira neprikladne principe

indukcije jer ne zna izraziti tvrdnju „predikat vrijedi za sve ugniježđene elemente”.

1 Lemma term_ind

2 : forall P : term Σ -> Prop,

3 (forall v, P (var_term v)) ->

4 (forall f args, (forall st, V.In st args -> P st) ->

5 P (TFunc f args)) ->

6 forall t : term Σ, P t.

Definicija 3. Skup svih varijabli terma 𝑡, s oznakom 𝑇𝑉(𝑡), konstruiramo rekurzivno na

način:

1. 𝑇𝑉(𝑣) ≔ ı𝑣# za varijablu 𝑣 i

2. 𝑇𝑉(𝑓(𝑡1, . . . , 𝑡𝑛)) ≔
⋃

1≤𝑖≤𝑛
𝑇𝑉(𝑡𝑖) za 𝑛-mjesni funkcijski simbol 𝑓 i terme 𝑡1, . . . , 𝑡𝑛.

1 Inductive TV : term -> var -> Prop :=

2 | TVVar : forall v, TV (var_term v) v

3 | TVFunc : forall f args v st,

4 V.In st args -> TV st v -> TV (TFunc f args) v.

5ili općenitije, vezivanju varijabli
6https://github.com/uds-psl/autosubst2
7ovdje vec

Definicija 4. Skup svih formula konstruiramo rekurzivno na način:

1. ako je 𝑄 (obični ili induktivni) predikatni simbol mjesnosti 𝑛 te su 𝑡1, . . . , 𝑡𝑛 termi,

onda je 𝑄(𝑡1, . . . , 𝑡𝑛) atomarna formula;

2. ako je 𝜑 formula, onda su ‵𝜑 i ∀𝜑 također formule;

3. ako su 𝜑 i 𝜓 formule, onda je 𝜑 → 𝜓 također formula.

1 Inductive formula : Set :=

2 | FPred (P : PredS Σ) : vec (term Σ) (pred_ar P) -> formula

3 | FIndPred (P : IndPredS Σ) : vec (term Σ) (indpred_ar P) -> formula

4 | FNeg : formula -> formula

5 | FImp : formula -> formula -> formula

6 | FAll : formula -> formula.

Ostale veznike definiramo kao sintaksne pokrate.

1 Definition FAnd (φ ψ : formula) : formula := FNeg (FImp φ (FNeg ψ)).

2 Definition FOr (φ ψ : formula) : formula := FImp (FNeg φ) ψ.

3 Definition FExist (φ : formula) : formula := FNeg (FAll (FNeg φ)).

Primjer 2. U proširenoj Peanovoj signaturi, svojstvo „svaki prirodan broj je paran ili

neparan” možemo izraziti formulom ∀𝑥,𝑁𝑎𝑡(𝑥)→ 𝐸𝑣𝑒𝑛(𝑥) ∨ 𝑂𝑑𝑑(𝑥).

1 Definition every_nat_is_even_or_odd : formula Σ__PA :=

2 FAll

3 (FImp

4 (FIndPred PA_Nat [var_term 0])

5 (FOr

6 (FIndPred PA_Even [var_term 0])

7 (FIndPred PA_Odd [var_term 0]))).

Ovdje vidimo učinak de Bruijnovog indeksiranja na zapis formula; umjesto eksplicitnog

navođenja varijable 𝑥 kod kvantifikacije, implicitno kvantificiramo po varijabli 0.

Definicija 5. Skup slobodnih varijabli formule 𝜑, s oznakom 𝐹𝑉(𝜑), konstruiramo re-

kurzivno na način:

1. 𝐹𝑉(𝑃(𝑢1, . . . , 𝑢𝑛)) ≔
⋃

1≤𝑖≤𝑛
𝑇𝑉(𝑢𝑖),

2. 𝐹𝑉(‵𝜑) ≔ 𝐹𝑉(𝜑),

3. 𝐹𝑉(𝜑 → 𝜓) ≔ 𝐹𝑉(𝜑) ∪ 𝐹𝑉(𝜓),

4. 𝐹𝑉(∀𝜑) ≔ ı𝑣 ∣ 𝑣 + 1 ∈ 𝐹𝑉(𝜑)#.

1 Inductive FV : formula -> var -> Prop :=

2 | FV_Pred : forall R args v st,

3 V.In st args -> TV st v -> FV (FPred R args) v

4 | FV_IndPred : forall R args v st,

5 V.In st args -> TV st v -> FV (FIndPred R args) v

6 | FV_Imp_l : forall F G v, FV F v -> FV (FImp F G) v

7 | FV_Imp_r : forall F G v, FV G v -> FV (FImp F G) v

8 | FV_Neg : forall F v, FV F v -> FV (FNeg F) v

9 | FV_All : forall F v, FV F (S v) -> FV (FAll F) v.

Definicija 6. Supstitucija je svaka funkcija iz skupa ℕ u skup svih terma. Supstituciju

𝜎možemo promatrati kao niz terma 𝑡0, 𝑡1, 𝑡2, . . . . Tada je pomaknuta supstitucija, s ozna-

kom 𝑡 ⋅ 𝜎, supstitucija koja odgovara nizu 𝑡, 𝑡0, 𝑡1, 𝑡2, . . . , za neki term 𝑡.

Domena supstitucijemože se rekurzivno proširiti na skup svih terma i skup svih formula.

1 Fixpoint subst_term (σ : var -> term) (t : term) : term :=

2 match t with

3 | var_term v => σ v

4 | TFunc f args => TFunc f (V.map (subst_term σ) args)

5 end.

1 Fixpoint subst_formula

2 (σ : var -> term Σ) (φ : formula )

3 : formula :=

4 match φ return formula with

5 | FPred P args => FPred P (V.map (subst_term σ) args)

6 | FIndPred P args => FIndPred P (V.map (subst_term σ) args)

7 | FNeg ψ => FNeg (subst_formula σ ψ)

8 | FImp ψ ξ => FImp (subst_formula σ ψ) (subst_formula σ ξ)

9 | FAll ψ => FAll (subst_formula (up_term_term σ) ψ)

10 end.

Ovdje funkcija up_term_term brine da supstitucija 𝜎 mijenja samo one varijable koje

nisu vezane. Pišemo 𝜑[𝜎] za primjenu supstitucije 𝜎 na formulu 𝜑. Često korištene

supstitucije formula su supstitucija varijable 𝑥 termom 𝑡 u formuli 𝜑, s oznakom 𝜑[𝑡∕𝑥],

te supstitucija svake varijable 𝑛 u formuli 𝜑 varijablom 𝑛+1, s oznakom 𝜑↑. Iste notacije

koristimo i za supstitucije na termima, listama terma i listama formula.

Konačno, potrebno je definirati sintaksu za indukciju. U Coqu su definicije induk-

tivnih propozicija proizvoljne do na ograničenje pozitivnosti, no radi jednostavnosti u

𝐹𝑂𝐿𝐼𝐷 su moguće samo induktivne definicije s atomarnim formulama, a pišemo ih u

stilu prirodne dedukcije:

𝑄1𝐮1. . .𝑄𝑛𝐮𝑛 𝑃1𝐯1. . .𝑃𝑚𝐯𝑚 .
𝑃𝐭

Ovdje su𝑄1, . . . , 𝑄𝑛 obični predikatni simboli, 𝑃1, . . . , 𝑃𝑚 i 𝑃 su induktivni predikatni sim-

boli, a podebljani znakovi predstavljaju 𝑛-torke terma, gdje je 𝑛mjesnost odgovarajućeg

predikata.

Definicija 7. Produkcija je uređena četvorka

1. liste parova običnih predikatnih simbola i 𝑛-torki terma odgovarajućih duljina,

2. liste parova induktivnih predikatnih simbola i𝑛-torki terma odgovarajućih duljina,

3. induktivnog predikatnog simbola 𝑃 mjesnosti𝑚 i

4. 𝑚-torke terma.

1 Record production :=

2 mkProd {

3 preds : list {P: PredS Σ & vec (term Σ) (pred_ar P)};

4 indpreds : list {P: IndPredS Σ & vec (term Σ) (indpred_ar P)};

5 indcons : IndPredS Σ;

6 indargs : vec (term Σ) (indpred_ar indcons);

7 }.

Prvi i drugi član četvorke zovemo premisama, a treći i četvrti konkluzijom. U ostatku

rada odabiremo neki podskup skupa svih produkcija koji zovemo skupom induktivnih

definicija, a označavamo s Φ.

1 Definition IndDefSet := production -> Prop.

U idućim primjerima definiramo produkcije za induktivne predikatne simbole u pro-

širenoj Peanovoj signaturi.

Primjer 3. Za broj kažemo da je prirodan ako je 0 ili je sljedbenik nekog prirodnog broja.

(PA_prod_N_zero)
𝑁𝑎𝑡(𝑜)

1 Definition PA_prod_N_zero : production Σ__PA.

2 refine (mkProd nil nil PA_Nat _).

3 refine [TFunc PA_zero []].

4 Defined.

𝑁𝑎𝑡(𝑥)
(PA_prod_N_succ)

𝑁𝑎𝑡(𝑠(𝑥))

1 Definition PA_prod_N_succ : production Σ__PA.

2 refine (mkProd nil _ PA_Nat _).

3 - refine (cons _ nil). exists PA_Nat; refine [var_term 0].

4 - refine [TFunc PA_succ [var_term 0]].

5 Defined.

Napomena. Uprimjeru 3 vidimo alternativni način definiranja uCoqu. Korisnik umjesto

definiranja Gallininih terma ulazi u način dokazivanja te naredbom refine postepeno

konstruira („profinjuje”) rezultatni term, ostavljajući „rupe” _ koje će popuniti kasnijim

primjenama naredbe refine. Konačni rezultat možemo vidjeti naredbom Print.

Print PA_prod_N_succ.

PA_prod_N_succ =

{|

preds := Datatypes.nil;

indpreds := (PA_Nat; [var_term 0]) :: Datatypes.nil;

indcons := PA_Nat;

indargs := [TFunc PA_succ [var_term 0]]

|}

: production

Primjer 4. Za broj kažemo da je paran ako je 0 ili je sljedbenik nekog neparnog broja.

(PA_prod_E_zero)
𝐸𝑣𝑒𝑛(𝑜)

1 Definition PA_prod_E_zero : production Σ__PA.

2 refine (mkProd nil nil PA_Even _).

3 refine [ TFunc PA_zero []].

4 Defined.

𝑂𝑑𝑑(𝑥)
(PA_prod_E_succ)

𝐸𝑣𝑒𝑛(𝑠(𝑥))

1 Definition PA_prod_E_succ : production Σ__PA.

2 refine (mkProd nil _ PA_Even _).

3 - refine (cons _ nil). exists PA_Odd; refine [var_term 0].

4 - refine [TFunc PA_succ [var_term 0]].

5 Defined.

Za broj kažemo da je neparan ako je sljedbenik nekog parnog broja.

𝐸𝑣𝑒𝑛(𝑥)
(PA_prod_O_succ)

𝑂𝑑𝑑(𝑠(𝑥))

1 Definition PA_prod_O_succ : production Σ__PA.

2 refine (mkProd nil _ PA_Odd _).

3 - refine (cons _ nil). exists PA_Even; refine [var_term 0].

4 - refine [TFunc PA_succ [var_term 0]].

5 Defined.

Skup prethodnih pet produkcija nazivamo skupom induktivnih definicija za proširenu

Peanovu signaturu, s oznakom Φ𝑃𝐴.

1 Inductive Φ__PA : production Σ__PA -> Prop :=

2 | ID_N_zero : Φ__PA PA_prod_N_zero

3 | ID_N_succ : Φ__PA PA_prod_N_succ

4 | ID_E_zero : Φ__PA PA_prod_E_zero

5 | ID_E_succ : Φ__PA PA_prod_E_succ

6 | ID_O_succ : Φ__PA PA_prod_O_succ.

2.2. Semantika

Definicija 8. Struktura prvog reda (kratko: struktura) je uređena četvorka skupa𝑀 koji

nazivamo nosačem te interpretacijŕa funkcijskih, običnih predikatnih i induktivnih predi-

katnih simbola. Funkcijski simboli mjesnosti 𝑛 interpretiraju se kao 𝑛-mjesne funkcije,

a predikatni simboli mjesnosti 𝑛 kao 𝑛-mjesne relacije na nosaču. Koristit ćemo ime

nosača kao sinonim za čitavu strukturu, a interpretacije označavati s 𝑓𝑀 odnosno 𝑃𝑀.

1 Structure structure := {

2 domain :> Set;

3 interpF (f : FuncS Σ) : vec domain (fun_ar f) -> domain;

4 interpP (P : PredS Σ) : vec domain (pred_ar P) -> Prop;

5 interpIP (P : IndPredS Σ) : vec domain (indpred_ar P) -> Prop;

6 }.

Primjer 5. Prije no što možemo definirati strukturu za Peanovu aritmetiku s induk-

tivnim predikatima, odnosno strukturu za signaturu Σ𝑃𝐴 i skup induktivnih definicija

Φ𝑃𝐴, potrebno je definirati odgovarajuće interpretacije induktivnih predikatnih simbola

u Coqu, u skladu s primjerima 3 i 4.

1 Inductive NAT : nat -> Prop :=

2 | NO : NAT 0

3 | NS : forall n, NAT n -> NAT (S n).

4

5 Inductive EVEN : nat -> Prop :=

6 | EO : EVEN 0

7 | ES : forall n, ODD n -> EVEN (S n)

8 with ODD : nat -> Prop :=

9 | OS : forall n, EVEN n -> ODD (S n).

Ovdje su predikati EVEN i ODD definirani mehanizmom simultane rekurzije.

Konačnomožemo definirati strukturu𝑀𝑃𝐴 s nosačemℕ te uobičajenom interpretacijom

funkcijskih i običnih predikatnih simbola. Induktivne predikatne simbole 𝑁𝑎𝑡, 𝐸𝑣𝑒𝑛 i

𝑂𝑑𝑑 interpretiramo redom Coqovim predikatima NAT, EVEN i ODD.

1 Definition M__PA : structure Σ__PA.

2 refine (Build_structure nat _ _ _).

3 - intros f; destruct f.

4 + intros. exact 0.

5 + intros n. exact (S (V.hd n)).

6 + intros xy. exact (V.hd xy + V.hd (V.tl xy)).

7 + intros xy. exact (V.hd xy * V.hd (V.tl xy)).

8 - intros P args; destruct P.

9 exact (V.hd args = V.hd (V.tl args)).

10 - intros P args; destruct P.

11 + exact (NAT (V.hd args)).

12 + exact (EVEN (V.hd args)).

13 + exact (ODD (V.hd args)).

14 Defined.

Napomena. U primjeru 5 koristimo taktiku exact koja je specijalizacija taktike refine,

a koristimo ju u zadnjem koraku konstrukcije terma. Nadalje, taktika intros x služi

uvođenju imena x u kontekst, a u našem je slučaju ekvivalentna taktici refine (fun x

=> _). Konačno, taktika destruct služi analizi po slučajevima, a u pozadini koristi pat-

ternmatching. Kako su interpretacije funkcijskih simbola funkcije koje primaju vektore,

koristimo Coqove funkcije V.hd i V.tl za dohvaćanje glave, odnosno repa vektora.

Definicija 9. Neka je𝑀 proizvoljna struktura. Okolina 𝜌 za𝑀 je proizvoljna funkcija

iz skupa varijabli (prirodnih brojeva) u nosač strukture.

1 Definition env := var -> M.

Okolina se može interpretirati kao niz 𝑑0, 𝑑1, 𝑑2, . . . . Tada je pomaknuta okolina, s ozna-

kom 𝑑 ⋅ 𝜌, niz 𝑑, 𝑑0, 𝑑1, 𝑑2, . . . za neki 𝑑 ∈ 𝑀. Proširenje domene okoline 𝜌 na skup svih

terma zovemo evaluacijom.

1 Fixpoint eval (ρ : env) (t : term Σ) : M :=

2 match t with

3 | var_term x => ρ x

4 | TFunc f args => interpF f (V.map (eval ρ) args)

5 end.

Pišemo 𝑡𝜌 za evaluaciju terma 𝑡 u okolini 𝜌. Istu notaciju koristimo i za evaluaciju 𝑛-torki

terma.

Definicija 10. Neka je𝑀 proizvoljna struktura te 𝜌 okolina za𝑀. Istinitost formule 𝜑 u

okolini 𝜌, s oznakom 𝜌 ⊨ 𝜑, definiramo rekurzivno na način:

1. ako je 𝑃 (obični ili induktivni) predikatni simbol mjesnosti 𝑛 te su 𝑢1, . . . , 𝑢𝑛 termi,

onda vrijedi 𝜌 ⊨ 𝑃(𝑢1, . . . , 𝑢𝑛) ako i samo ako vrijedi 𝑃
𝑀(𝑢

𝜌

1 , . . . , 𝑢
𝜌
𝑛),

2. vrijedi 𝜌 ⊨ ‵𝜑 ako i samo ako ne vrijedi 𝜌 ⊨ 𝜑 (što još pišemo 𝜌 ⊭ 𝜑),

3. vrijedi 𝜌 ⊨ 𝜑 → 𝜓 ako i samo ako vrijedi 𝜌 ⊭ 𝜑 ili 𝜌 ⊨ 𝜓,

4. vrijedi 𝜌 ⊨ ∀𝜑 ako i samo ako za sve 𝑑 ∈ 𝑀 vrijedi 𝑑 ⋅ 𝜌 ⊨ 𝜑.

1 Fixpoint Sat (ρ : env M) (F : formula Σ) : Prop :=

2 match F with

3 | FPred P args => interpP P (V.map (eval ρ) args)

4 | FIndPred P args => interpIP P (V.map (eval ρ) args)

5 | FNeg G => ~ Sat ρ G

6 | FImp F G => Sat ρ F -> Sat ρ G

7 | FAll G => forall d, Sat (d .: ρ) G

8 end.

Primjer 6. Formula ∀𝑥,𝑁𝑎𝑡(𝑥) → 𝐸𝑣𝑒𝑛(𝑥) ∨ 𝑂𝑑𝑑(𝑥) je istinita u strukturi 𝑀𝑃𝐴 bez

obzira na njenu okolinu.

1 Lemma every_nat_is_even_or_odd_Sat :

2 forall (ρ : env M__PA), ρ ⊨ every_nat_is_even_or_odd.

Lema 1. Neka su 𝜑, 𝜎,𝑀 i 𝜌 redom proizvoljna formula, supstitucija, struktura i okolina

za𝑀. Tada vrijedi 𝜌 ⊨ 𝜑[𝜎] ako i samo ako vrijedi (𝑡 ↦ 𝑡𝜌) ◦𝜎 ⊨ 𝜑.

1 Lemma strong_form_subst_sanity2 :

2 forall (Σ : signature) (φ : formula Σ) (σ : var -> term Σ)

3 (M : structure Σ) (ρ : env M),

4 ρ ⊨ (subst_formula σ φ) <-> (σ >> eval ρ) ⊨ φ.

Kompoziciju supstitucije 𝜎 i evaluacije 𝑡 ↦ 𝑡𝜌 možemo nazvati semantičkom supstituci-

jom jer prvo provodi sintaksnu supstituciju 𝜎 nakon čega provodi evaluaciju. Tada mo-

žemo neformalno reći da sintaksna i semantička supstitucija komutiraju pod relacijom

istinitosti.

2.3. Standardni modeli

Želimo ograničiti semantička razmatranja na samo one strukture koje „imaju smisla” za

induktivne predikate. Prisjetimo se, predikatni simbol 𝑃 mjesnosti 𝑛 interpretira se na

strukturi 𝑀 podskupom skupa 𝑀𝑛. Indukciju smatramo dokazivanjem u razinama pa

ima smisla promatrati razine interpretacije induktivnog predikata, gdje je nulta razina

prazan skup, a svaku iduću razinu konstruiramo pomoću produkcija induktivnog skupa

definicija i prethodnih razina. Tako je prva razina onaj podskup kojeg možemo dobiti

najviše jednom primjenom produkcija, druga je razina onaj podskup kojeg možemo do-

biti pomoću najviše dviju primjena produkcija, i tako dalje. Na taj se način, korak po ko-

rak, gradi smislena interpretacija induktivnih predikata. Zbog mogućih međuovisnosti

induktivnih predikata, razine interpretacije moramo definirati simultano. Ovaj odjeljak

posvećujemo formalizaciji ovih pojmova.

Definicija 11. Neka je𝑀 proizvoljna struktura te neka je 𝑝𝑟 proizvoljna produkcija in-

duktivnog skupa definicija Φ, primjerice:

𝑄1𝐮1. . .𝑄𝑛𝐮𝑛 𝑃1𝐯1. . .𝑃𝑚𝐯𝑚 .
𝑃 𝐭

Neka je 𝐼 proizvoljna interpretacija induktivnih predikatnih simbola. Tada definiramo

𝜑𝑝𝑟(𝐼) kao skup svih |𝑃|-torki 𝐝 elemenata nosača𝑀 za koje postoji okolina 𝜌 za𝑀 takva

da:

• za sve 𝑖 ∈ ı1, . . . , 𝑛# vrijedi 𝐮𝜌𝑖 ∈ 𝑄𝑀
𝑖 ,

• za sve 𝑗 ∈ ı1, . . . , 𝑚# vrijedi 𝐯𝜌𝑗 ∈ 𝑓(𝑃𝑗) i

• 𝐝 = 𝐭𝜌.

1 Definition φ_pr

2 (pr : production Σ)

3 (interp : InterpInd)

4 (ds : vec M (indpred_ar (indcons pr)))

5 : Prop :=

6 exists (ρ : env M),

7 (forall Q us, List.In (Q; us) (preds pr) ->

8 interpP Q (V.map (eval ρ) us)) /\

9 (forall P ts, List.In (P; ts) (indpreds pr) ->

10 interp P (V.map (eval ρ) ts)) /\

11 ds = V.map (eval ρ) (indargs pr).

Operator 𝜑𝑝𝑟 je formalizacija ideje primjene produkcije. Nadalje, potrebno je definirati

operator koji će uzeti u obzir sve produkcije koje se odnose na 𝑃. Definiramo 𝜑𝑃(𝑓) kao

uniju svih 𝜑𝑝𝑟′(𝑓) gdje je 𝑝𝑟
′ produkcija u kojoj se 𝑃 javlja u konkluziji.

1 Definition φ_P

2 (P : IndPredS Σ)

3 (interp : InterpInd)

4 : vec M (indpred_ar P) -> Prop.

5 refine (fun ds => _).

6 refine (@ex (production Σ) (fun pr => _)).

7 refine (@ex (P = indcons pr /\ Φ pr) (fun '(conj Heq HΦ) => _)).

8 rewrite Heq in ds.

9 exact (φ_pr pr interp ds).

10 Defined.

Konačno, definiramo operator skupa definicija 𝜑Φ kao preslikavanje koje svakom induk-

tivnom predikatnom simbolu 𝑃 pridružuje skup 𝜑𝑃(𝑓).

1 Definition φ_Φ (interp : InterpInd) : InterpInd :=

2 fun P => φ_P P interp.

Operator 𝜑Φ omogućuje simultanu primjenu produkcija.

Napomena. Kako je funkcija 𝑓 bila uvedena na samom početku prethodne definicije, u

stvari definicija operatora 𝜑Φ glasi 𝜑Φ(𝑓)(𝑃) ≔ 𝜑𝑃(𝑓).

Primjer 7. Ilustrirajmo operator𝜑Φ na strukturi𝑀𝑃𝐴 i skupu induktivnih definicijaΦ𝑃𝐴.

Prisjetimo se, u proširenoj Peanovoj signaturi imamo tri induktivna predikatna simbola;

𝑁𝑎𝑡, 𝐸𝑣𝑒𝑛 i 𝑂𝑑𝑑, i svi su oni jednomjesni. Dakle, domena i kodomena operatora 𝜑Φ

u ovom primjeru je skup 𝒫(ℕ)
3
, gdje prva, druga i treća projekcija odgovaraju redom

27



interpretacijama predikata 𝑁𝑎𝑡, 𝐸𝑣𝑒𝑛 i 𝑂𝑑𝑑. Nadalje, vrijedi:

𝜑Φ𝑃𝐴,𝑀𝑃𝐴
(𝑁,𝐸,𝑂) = (ı0# ∪ ı𝑛 + 1 ∣ 𝑛 ∈ 𝑁#, ı0# ∪ ı𝑜 + 1 ∣ 𝑜 ∈ 𝑂#, ı𝑒 + 1 ∣ 𝑒 ∈ 𝐸#).

Prema notaciji u prethodnoj napomeni, funkcija 𝑓 je ovdje preslikavanje

(𝑁𝑎𝑡 ↦ 𝑁,𝐸𝑣𝑒𝑛 ↦ 𝐸,𝑂𝑑𝑑 ↦ 𝑂)

kojemožemo reprezentirati uređenom trojkom (𝑁,𝐸,𝑂). Indeksiranje induktivnim pre-

dikatnim simbolom odgovara prethodno navedenim projekcijama pa vrijedi primjerice:

𝜑Φ𝑃𝐴,𝑀𝑃𝐴
(𝑁,𝐸,𝑂)(𝑁𝑎𝑡) = ı0# ∪ ı𝑛 + 1 ∣ 𝑛 ∈ 𝑁#.

Definicija 12. Neka su 𝑈 i 𝑉 proizvoljne interpretacije induktivnih predikatnih sim-

bola. Kažemo da je interpretacija 𝑈 podinterpretacija interpretacije 𝑉 i pišemo 𝑈 ⊑ 𝑉

ako za svaki induktivni predikatni simbol 𝑃 vrijedi 𝑈(𝑃) ⊆ 𝑉(𝑃).

1 Definition subinterp (U V : InterpInd) := forall P v, U P v -> V P v.

2 Notation "U ⊑ V" := (subinterp U V) (at level 11).

Propozicija 1. Operator 𝜑Φ je monoton.

1 Definition monotone (F : InterpInd -> InterpInd) :=

2 forall U V : InterpInd, U ⊑ V -> F U ⊑ F V.

3

4 Proposition φ_Φ_monotone : monotone φ_Φ.

Definicija 13. Neka je 𝑀 proizvoljna struktura. Definiramo aproksimaciju skupa in-

duktivnih definicija Φ razine 𝛼 ∈ ℕ, s oznakom 𝜑𝛼Φ, rekurzivno na način:

1. 𝜑0Φ(𝑃) ≔ ∅ i

2. 𝜑𝛼+1Φ ≔ 𝜑Φ(𝜑
𝛼
Φ).

Neformalno možemo reći da operator 𝜑Φ poboljšava aproksimaciju prethodne razine.

1 Fixpoint φ_Φ_n (α : nat) : InterpInd :=

2 match α with

3 | 0 => fun _ _ => False

4 | S α => φ_Φ (φ_Φ_n α)

5 end.

Tada je aproksimant induktivnog predikatnog simbola 𝑃 razine 𝛼 upravo 𝜑𝛼Φ(𝑃).

Napomena. Brotherston je definirao aproksimaciju razine 𝛼 za pojedini induktivni pre-

dikatni simbol kao uniju poboljšanih aproksimacija svih nižih razina. Takva je definicija

ekvivalentna našoj.

Lema 2. Za svaki prirodni broj 𝛼 i induktivni predikatni simbol 𝑃 vrijedi

𝜑𝛼Φ(𝑃) =
⋃

𝛽<𝛼

𝜑Φ(𝜑
𝛽

Φ(𝑃)).

1 Lemma φ_Φ_n_characterization : forall α P v,

2 φ_Φ_n α P v <-> exists β, β < α /\ φ_Φ (φ_Φ_n β) P v.

Primjer 8. Za svaki prirodni broj 𝛼, a u skladu s primjerom 7, aproksimant razine 𝛼

induktivnog predikatnog simbola 𝑁𝑎𝑡 je skup svih prirodnih brojeva strogo manjih od

𝛼, odnosno vrijedi:

𝜑𝛼Φ𝑃𝐴,𝑀𝑃𝐴
(𝑁𝑎𝑡) = ı0, 1, . . . , 𝛼 − 1#.

1 Lemma approximants_of_PA_Nat : forall α n,

2 φ_Φ_n Φ__PA M__PA α PA_Nat [n] <-> n < α.

Definicija 14. Aproksimacija razine 𝜔, s oznakom 𝜑𝜔Φ, je (za svaki pojedini induktivni

predikatni simbol) unija aproksimacija razina manjih od 𝜔.

1 Definition φ_Φ_ω : InterpInd := fun P v => exists α, φ_Φ_n α P v.

Primjer 9. Aproksimant razine 𝜔 induktivnog predikatnog simbola 𝑁𝑎𝑡 je cijeli skup

prirodnih brojeva ℕ.

1 Lemma NAT_φ_Φ_ω: forall n, φ_Φ_ω Φ__PA M__PA PA_Nat [n].

Definicija 15. Neka je 𝑈 proizvoljna interpretacija induktivnih predikatnih simbola.

Kažemo da je interpretacija𝑈 prefiksna točka ako je poboljšanje interpretacije𝑈 podin-

terpretacija od𝑈, odnosno vrijedi𝜑Φ(𝑈) ⊑ 𝑈. Za interpretaciju induktivnih predikatnih

simbola 𝑉 kažemo da je najmanja prefiksna točka ako je 𝑉 prefiksna točka te je podin-

terpretacija svake druge prefiksne točke.

1 Definition prefixed (U : InterpInd) := φ_Φ U ⊑ U.

2 Definition least (P : InterpInd -> Prop) (U : InterpInd) :=

3 P U /\ (forall V, P V -> U ⊑ V).

Lema 3. Aproksimacija razine 𝜔 je najmanja prefiksna točka operatora 𝜑Φ.

1 Lemma φ_Φ_ω_least_prefixed : least prefixed φ_Φ_ω.

Definicija 16. Kažemo da je struktura𝑀 standardni model za Φ ako interpretira svaki

induktivni predikatni simbol njegovim aproksimantom razine 𝜔.

1 Definition standard_model (Φ: IndDefSet Σ) (M : structure Σ) : Prop :=

2 forall (P : IndPredS Σ) ts, interpIP P ts <-> φ_Φ_ω Φ M P ts.

Prema lemi 3, standardni model je najmanja struktura koja ima smisla za zadani skup

induktivnih definicija.

Propozicija 2. Struktura𝑀𝑃𝐴 je standardni model za skup induktivnih definicija Φ𝑃𝐴.

1 Lemma standard_model__PA : standard_model Φ__PA M__PA.

3. Sistem sekvenata s induktivnim

definicijama

Cilj je ovog odjeljka definirati dokazni sustav 𝐿𝐾𝐼𝐷 za logiku 𝐹𝑂𝐿𝐼𝐷. Ovaj dokazni sus-

tav temelji se na Gentzenovu računu sekvenata, a proširen je lijevim i desnim pravilima

za induktivne predikate. Prije no što definiramo 𝐿𝐾𝐼𝐷, potrebno je definirati pojam sek-

vente i međusobne zavisnosti induktivnih predikata.

Definicija 17. Neka su Γ i ∆ proizvoljne liste formula. Tada je sekventa uređeni par

(Γ,∆), a označavamo ju s Γ ⊢ ∆.

1 Inductive sequent : Set :=

2 | mkSeq (Γ Δ : list (formula Σ)).

Sekventom Γ ⊢ ∆ konceptualno tvrdimo da istinitost svake tvrdnje u Γ povlači istinitost

neke tvrdnje u ∆. Kažemo da je sekventa Γ ⊢ ∆ dokaziva, ili da Γ izvodi ∆, ako u sustavu

𝐿𝐾𝐼𝐷 postoji dokaz za Γ ⊢ ∆. Neformalno možemo reći da je dokaz sekvente Γ ⊢ ∆

konačno stablo u kojem je sekventa Γ ⊢ ∆ korijen te svaki čvor stabla slijedi iz svoje

djece u skladu s pravilima izvoda sustava 𝐿𝐾𝐼𝐷. Kasnije ćemo precizirati pojam dokaza.

Definicija 18. Neka su 𝑃𝑖 i 𝑃𝑗 proizvoljni induktivni predikatni simboli. Kažemo da

je simbol 𝑃𝑖 u relaciji 𝑃𝑟𝑒𝑚 sa simbolom 𝑃𝑗 ako postoji produkcija u skupu induktivnih

definicijaΦ takva da se simbol 𝑃𝑖 javlja u njenoj konkluziji te se simbol 𝑃𝑗 javlja u njenim

premisama.

1 Definition Prem (Pi Pj : IndPredS Σ) :=

2 exists pr, Φ pr /\

3 indcons pr = Pi /\

4 exists ts, List.In (Pj; ts) (indpreds pr).

Definicija 19. Definiramo relaciju 𝑃𝑟𝑒𝑚∗ kao refleksivno i tranzitivno zatvorenje rela-

cije 𝑃𝑟𝑒𝑚 na skupu induktivnih predikatnih simbola.

1 Definition Prem_star := clos_refl_trans (IndPredS Σ) Prem.

Za induktivne predikatne simbole 𝑃 i 𝑄 kažemo da su međusobno zavisni ako vrijedi

𝑃𝑟𝑒𝑚∗(𝑃,𝑄) i 𝑃𝑟𝑒𝑚∗(𝑄, 𝑃).

1 Definition mutually_dependent (P Q : IndPredS Σ) :=

2 Prem_star P Q /\ Prem_star Q P.

Za međusobno zavisne induktivne predikatne simbole kažemo još da su definirani si-

multanom rekurzijom. Simultana rekurzija se koristi za formalnu definicijumeđusobno

zavisnih induktivnih predikata i u mnogo složenijim sustavima od 𝐹𝑂𝐿𝐼𝐷; primjerice,

Coqovi predikati EVEN i ODD iz primjera 5 definirani su simultanom rekurzijom.

Primjer 10. U skupu induktivnih definicija Φ𝑃𝐴, induktivni predikatni simboli 𝐸𝑣𝑒𝑛 i

𝑂𝑑𝑑 su međusobno zavisni.

1 Example mut_dep_E_O :

2 mutually_dependent Φ__PA PA_Even PA_Odd.

S druge strane, induktivni predikatni simboli 𝑁𝑎𝑡 i 𝐸𝑣𝑒𝑛 nisumeđusobno zavisni.

1 Example not_mut_dep_N_E :

2 ~ mutually_dependent Φ__PA PA_Nat PA_Even.

Lema 4. Međusobna zavisnost je relacija ekvivalencije na skupu induktivnih predikat-

nih simbola.

1 Lemma mutually_dependent_equiv : Equivalence mutually_dependent.

Sustav 𝐿𝐾𝐼𝐷 sastoji se od četiri vrste pravila izvoda: strukturalna, propozicijska, pra-

vila za kvantifikatore i pravila za induktivne predikate. Navest ćemo ih tim redom, a

definirati kroz više blokova Coqova koda. Napominjemo da se sustav 𝐿𝐾𝐼𝐷 razlikuje

od Gentzenova sustava 𝐿𝐾 na nekoliko mjesta, zbog implementacijskih detalja te zbog

induktivnih definicija. Za neko pravilo izvoda kažemo da je dopustivo ako sve što mo-

žemo dokazati (pod nekim pretpostavkama) korištenjem tog pravila, možemo dokazati

pod istim pretpostavkama i bez njegova korištenja.

1 Inductive LKID : sequent -> Prop := (* nastavlja se *)

3.1. Strukturalna pravila

Strukturalna su pravila navedena na prikazu 1, a služe baratanju strukturom sekvente.

Iako je sekventa implementirana kao par lista formula, te liste se ponašaju kao skupovi u

strukturalnim pravilima jer za sadržanost jedne liste u drugoj koristimo Coqov predikat

incl koji se oslanja na predikat In. Posljedično, pravilo slabljenja 𝑊𝑘 povlači pravila

permutacije i kontrakcije.

Γ ∩ ∆ ≠ ∅
(𝐴𝑥)

Γ ⊢ ∆

Γ′ ⊢ ∆′ Γ′ ⊆ Γ ∆′ ⊆ ∆
(𝑊𝑘)

Γ ⊆ ∆

Γ ⊢ 𝜑,∆ 𝜑,Γ ⊢ ∆
(𝐶𝑢𝑡)

Γ ⊢ ∆

Γ ⊢ ∆ (𝑆𝑢𝑏𝑠𝑡)
Γ[𝜎] ⊢ ∆[𝜎]

Prikaz 1: Strukturalna pravila sustava 𝐿𝐾𝐼𝐷.

2 (* Strukturalna pravila. *)

3 | Ax : forall Γ Δ φ, In φ Γ -> In φ Δ -> LKID (Γ ⊢ Δ)
4 | Wk : forall Γ' Δ' Γ Δ,

5 Γ' ⊆ Γ ->

6 Δ' ⊆ Δ ->

7 LKID (Γ' ⊢ Δ') ->

8 LKID (Γ ⊢ Δ)
9 | Cut : forall Γ Δ φ,

10 LKID (Γ ⊢ φ :: Δ) ->

11 LKID (φ :: Γ ⊢ Δ) ->

12 LKID (Γ ⊢ Δ)
13 | Subst : forall Γ Δ,

14 LKID (Γ ⊢ Δ) ->

15 forall σ, LKID (map (subst_formula σ) Γ ⊢ map (subst_formula σ) Δ)

Primjer 11. Prošireno pravilo aksioma je dopustivo.

(𝐴𝑥𝐸𝑥𝑡𝑒𝑛𝑑𝑒𝑑)
𝜑,Γ ⊢ 𝜑,∆

Lemma AxExtended : forall Γ Δ φ, LKID (φ :: Γ ⊢ φ :: Δ).

Nadalje, pravilo permutacije je dopustivo.

Γ′ ⊢ ∆′ Γ′ ∼ Γ ∆′ ∼ ∆ (𝑃𝑒𝑟𝑚)
Γ ⊢ ∆

Ovdje znakom ∼ označujemo relaciju permutacije na listama.

Lemma Perm : forall Γ' Δ' Γ Δ,

Permutation Γ' Γ ->

Permutation Δ' Δ ->

LKID (Γ' ⊢ Δ') ->

LKID (Γ ⊢ Δ).

Konačno, lijeva i desna pravila kontrakcije su dopustiva.

𝜑, 𝜑,Γ ⊢ ∆
(𝐶𝑜𝑛𝑡𝑟𝐿)

𝜑,Γ ⊢ ∆

Γ ⊢ 𝜑, 𝜑,∆
(𝐶𝑜𝑛𝑡𝑟𝑅)

Γ ⊢ 𝜑,∆

Lemma ContrL : forall Γ Δ φ,

LKID (φ :: φ :: Γ ⊢ Δ) -> LKID (φ :: Γ ⊢ Δ).
Lemma ContrR : forall Γ Δ φ,

LKID (Γ ⊢ φ :: φ :: Δ) -> LKID (Γ ⊢ φ :: Δ).

3.2. Propozicijska pravila

Propozicijska su pravila navedena na prikazu 2, a služe rasuđivanju s implikacijama i

negacijama. Kao što smo rekli, ostale propozicijske veznike shvaćamo kao pokrate. Ova

su pravila identična u sustavima 𝐿𝐾 i 𝐿𝐾𝐼𝐷.

Γ ⊢ 𝜑,∆
(𝑁𝑒𝑔𝐿)

‵𝜑,Γ ⊢ ∆

𝜑,Γ ⊢ ∆
(𝑁𝑒𝑔𝑅)

Γ ⊢ ‵𝜑,∆

Γ ⊢ 𝜑,∆ 𝜓,Γ ⊢ ∆
(𝐼𝑚𝑝𝐿)

𝜑 → 𝜓,Γ ⊢ ∆

𝜑,Γ ⊢ 𝜓,∆
(𝐼𝑚𝑝𝑅)

Γ ⊢ 𝜑 → 𝜓,∆

Prikaz 2: Propozicijska pravila sustava 𝐿𝐾𝐼𝐷.

16 (* Propozicijska pravila. *)

17 | NegL : forall Γ Δ φ, LKID (Γ ⊢ φ :: Δ) -> LKID (FNeg φ :: Γ ⊢ Δ)
18 | NegR : forall Γ Δ φ, LKID (φ :: Γ ⊢ Δ) -> LKID (Γ ⊢ FNeg φ :: Δ)

19 | ImpL : forall Γ Δ φ ψ,

20 LKID (Γ ⊢ φ :: Δ) -> LKID (ψ :: Γ ⊢ Δ) ->

21 LKID (FImp φ ψ :: Γ ⊢ Δ)
22 | ImpR : forall Γ Δ φ ψ,

23 LKID (φ :: Γ ⊢ ψ :: Δ) -> LKID (Γ ⊢ (FImp φ ψ) :: Δ)

Primjer 12. Desno pravilo za konjunkciju je dopustivo.

Γ ⊢ 𝜑,∆ Γ ⊢ 𝜓,∆
(𝐴𝑛𝑑𝑅)

Γ ⊢ 𝜑 ∧ 𝜓,∆

Lemma AndR : forall Γ Δ φ ψ,

LKID (Γ ⊢ φ :: Δ) -> LKID (Γ ⊢ ψ :: Δ) ->

LKID (Γ ⊢ FAnd φ ψ :: Δ).

Lijevo pravilo za disjunkciju je dopustivo.

𝜑,Γ ⊢ ∆ 𝜓,Γ ⊢ ∆
(𝑂𝑟𝐿)

𝜑 ∨ 𝜓,Γ ⊢ ∆

Lemma OrL : forall Γ Δ φ ψ,

LKID (φ :: Γ ⊢ Δ) -> LKID (ψ :: Γ ⊢ Δ) ->

LKID (FOr φ ψ :: Γ ⊢ Δ).

3.3. Pravila za kvantifikatore

Pravila za kvantifikatore navedena su na prikazu 3, a odnose se na univerzalnu kvantifi-

kaciju. Ova se pravila razlikuju od analognih u sustavu 𝐿𝐾 zbog korištenja de Bruijnovih

indeksa kod kvantifikacije. U lijevom pravilu koristimo supstituciju 𝑡 ⋅ 𝜎𝑖𝑑 u formuli 𝜑,

gdje je 𝜎𝑖𝑑 supstitucija identiteta, kako bismo iskazali da se term 𝑡 javlja na mjestu neke

varijable 𝑥. Za desno pravilo, primijetimo da se varijabla 0 ne javlja u listama formula

Γ↑ i ∆↑ zbog pomicanja. Kako implicitno kvantificiramo po varijabli 0, a ona se javlja

(eventualno) samo u formuli 𝜑, smijemo univerzalno kvantificirati formulu 𝜑.

𝜑[𝑡 ⋅ 𝜎𝑖𝑑],Γ ⊢ ∆
(𝐴𝑙𝑙𝐿)

∀𝜑,Γ ⊢ ∆

Γ↑ ⊢ 𝜑,∆↑

(𝐴𝑙𝑙𝑅)
Γ ⊢ ∀𝜑,∆

Prikaz 3: Pravila za kvantifikatore sustava 𝐿𝐾𝐼𝐷.

24 (* Kvantifikatorska pravila. *)

25 | AllL : forall Γ Δ φ t,

26 LKID (subst_formula (t .: ids) φ :: Γ ⊢ Δ) ->

27 LKID (FAll φ :: Γ ⊢ Δ)
28 | AllR : forall Γ Δ φ,

29 LKID (shift_formulas Γ ⊢ φ :: shift_formulas Δ) ->

30 LKID (Γ ⊢ (FAll φ) :: Δ)

Primjer 13. Lijevo pravilo za egzistencijalnu kvantifikaciju je dopustivo.

𝜑,Γ↑ ⊢ ∆↑

(𝐸𝑥𝑖𝑠𝑡𝐿)
∃𝜑,Γ ⊢ ∆

Lemma ExistL : forall Γ Δ φ,

LKID (φ :: shift_formulas Γ ⊢ shift_formulas Δ) ->

LKID (FExist φ :: Γ ⊢ Δ).

Preostaje nam definicija pravila za induktivne predikatne simbole, gdje ćemo prvo za

svaki induktivni predikatni simbol definirati lijevo pravilo koje nazivamo pravilo induk-

cije, a zatim ćemo za svaku produkciju definirati desno pravilo koje nazivamo produkcij-

sko pravilo.

3.4. Pravila indukcije

Neka je 𝑃𝑗 proizvoljni induktivan predikatni simbol za koji definiramo pravilo indukcije.

Svakom induktivnom predikatnom simbolu 𝑃𝑖 dodjeljujemo proizvoljnu |𝑃𝑖|-torku in-
duktivnih varijabli 𝐳𝑖 te proizvoljnu pretpostavku indukcije 𝐺𝑖 takvu da vrijedi 𝐺𝑖 = 𝑃𝑖(𝐳𝑖)

ako 𝑃𝑖 i 𝑃𝑗 nisu međusobno zavisni. Sada pravilo indukcije za induktivan predikatni

simbol 𝑃𝑗 glasi:

male premise 𝐺𝑗[𝐮∕𝐳𝑖],Γ ⊢ ∆
(𝑃𝑗 𝐼𝑛𝑑),𝑃𝑗𝐮,Γ ⊢ ∆

gdje sekventu𝐺𝑗[𝐮∕𝐳𝑖],Γ ⊢ ∆ zovemo velikom premisom, a za svaku produkciju u skupu

induktivnih definicija Φ u čijoj se konkluziji javlja induktivni predikatni simbol 𝑃𝑖 me-

đusobno zavisan sa simbolom 𝑃𝑗, primjerice:

𝑄1𝐮1. . .𝑄𝑛𝐮𝑛 𝑃𝑖1𝐯𝑖1 . . .𝑃𝑖𝑚𝐯𝑖𝑚 ,
𝑃𝑖𝐭

postojimala premisa oblika:

𝑄1𝐮
↑𝑘
1 , . . . , 𝑄𝑛𝐮

↑𝑘
𝑛 , 𝐺𝑖1

[𝐯↑𝑘𝑖1 ∕𝐳𝑖1], . . . , 𝐺𝑖𝑚
[𝐯↑𝑘𝑖𝑚 ∕𝐳𝑖𝑚],Γ ⊢ 𝐺𝑖[𝐭

↑𝑘∕𝐳𝑖],∆

gdje je 𝑘 neki prirodan broj takav da se nijedan 𝑘′ ≥ 𝑘 ne pojavljuje u listama formula Γ i

∆ te formuli 𝑃𝑗𝐮 (takav postoji jer se u tih konačnomnogo formula sveukupno pojavljuje

samo konačno mnogo varijabli), a s ↑𝑘 označavamo supstituciju koja svaku varijablu 𝑛

zamjenjuje varijablom 𝑛 + 𝑘.

Ilustrirajmo pravilo indukcije primjerima u Peanovoj aritmetici. Kako se pravila za

induktivne predikate sustava 𝐿𝐾𝐼𝐷 odnose na proizvoljni skup induktivnih definicijaΦ,

označimo s 𝐿𝐾𝐼𝐷𝑃𝐴 sustav koji se odnosi na skup induktivnih definicija Φ𝑃𝐴.

Definition LKID__PA := LKID Φ__PA.

Primjer 14. Pravilo indukcije simbola 𝑁𝑎𝑡 u sustavu 𝐿𝐾𝐼𝐷𝑃𝐴 glasi:

Γ ⊢ 𝐺(𝑜),∆ 𝐺(𝑥),Γ ⊢ 𝐺(𝑠(𝑥)),∆ 𝐺(𝑡),Γ ⊢ ∆
(𝑁𝑎𝑡𝐼𝑛𝑑),

𝑁𝑎𝑡(𝑡),Γ ⊢ ∆

gdje je G pretpostavka indukcije vezana uz induktivni predikatni simbol 𝑁𝑎𝑡, a 𝑥 je va-

rijabla koja se ne javlja u Γ, ∆ niti u 𝑁𝑎𝑡(𝑡).

Primjer 15. Pravilo indukcije simbola 𝐸𝑣𝑒𝑛 u sustavu 𝐿𝐾𝐼𝐷𝑃𝐴 glasi:

Γ ⊢ 𝐺𝐸(𝑜),∆ Γ, 𝐺𝐸(𝑥) ⊢ 𝐺𝑂(𝑠(𝑥)),∆ Γ, 𝐺𝑂(𝑥) ⊢ 𝐺𝐸(𝑠(𝑥)),∆ Γ, 𝐺𝐸(𝑡) ⊢ ∆
(𝐸𝑣𝑒𝑛𝐼𝑛𝑑),

Γ, 𝐸𝑣𝑒𝑛(𝑡) ⊢ ∆

gdje su 𝐺𝐸 i 𝐺𝑂 pretpostavke indukcije vezane uz induktivne predikatne simbole 𝐸𝑣𝑒𝑛 i

𝑂𝑑𝑑, a 𝑥 je varijabla koja se ne javlja u Γ, ∆ niti u 𝐸𝑣𝑒𝑛(𝑡).

31 (* Pravila indukcije. *)

32 | Ind : forall Γ Δ

33 (Pj : IndPredS Σ) (u : vec _ (indpred_ar Pj))

34 (z_i : forall P, vec var (indpred_ar P))

35 (z_i_nodup : forall P, VecNoDup (z_i P))

36 (G_i : IndPredS Σ -> formula Σ)

37 (HG2 : forall Pi, ~mutually_dependent Pi Pj ->

38 G_i Pi = FIndPred

39 Pi

40 (V.map var_term (z_i Pi))),

41 let maxΓ := max_fold (map some_var_not_in_formula Γ) in

42 let maxΔ := max_fold (map some_var_not_in_formula Δ) in

43 let maxP := some_var_not_in_formula (FIndPred Pj u) in

44 let shift_factor := max maxP (max maxΓ maxΔ) in

45 let Fj := subst_formula

46 (finite_subst (z_i Pj) u)

47 (G_i Pj) in

48 let minor_premises :=

49 (forall pr (HΦ:Φ pr) (Hdep: mutually_dependent (indcons pr) Pj),

50 let Qs := shift_formulas_by

51 shift_factor

52 (FPreds_from_preds (preds pr)) in

53 let Gs := map (fun '(P; args) =>

54 let shifted_args :=

55 V.map

56 (shift_term_by shift_factor)

57 args in

58 let σ :=

59 finite_subst

60 (z_i P)

61 (shifted_args) in

62 let G := G_i P in

63 subst_formula σ G)

64 (indpreds pr) in

65 let Pi := indcons pr in

66 let ty := V.map

67 (shift_term_by shift_factor)

68 (indargs pr) in

69 let Fi := subst_formula

70 (finite_subst (z_i Pi) ty)

71 (G_i Pi) in

72 LKID (Qs ++ Gs ++ Γ ⊢ Fi :: Δ))

73 in

74 minor_premises ->

75 LKID (Fj :: Γ ⊢ Δ) ->

76 LKID (FIndPred Pj u :: Γ ⊢ Δ)

3.5. Produkcijska pravila

Neka je 𝑝𝑟 proizvoljna produkcija za koju definiramo produkcijsko pravilo. Možemo

pretpostaviti da je 𝑝𝑟 oblika:

𝑄1𝐮1. . .𝑄𝑛𝐮𝑛 𝑃1𝐯1. . .𝑃𝑚𝐯𝑚 .
𝑃𝐭

Tada produkcijsko pravilo za produkciju 𝑝𝑟 (i proizvoljnu supstituciju 𝜎) glasi:

Γ ⊢ 𝑄1𝐮1[𝜎],∆ . . . Γ ⊢ 𝑄𝑛𝐮𝑛[𝜎],∆ Γ ⊢ 𝑃1𝐯1[𝜎],∆ . . . Γ ⊢ 𝑃𝑚𝐯𝑚[𝜎],∆ (𝑝𝑟𝑃𝑟𝑜𝑑).
Γ ⊢ 𝑃𝐭[𝜎],∆

77 (* Produkcijska pravila. *)

78 | Prod : forall Γ Δ pr σ,

79 Φ pr ->

80 (forall Q us,

81 In (Q; us) (preds pr) ->

82 LKID (Γ ⊢ (FPred Q (V.map (subst_term σ) us) :: Δ))) ->

83 (forall P ts,

84 In (P; ts) (indpreds pr) ->

85 LKID (Γ ⊢ (FIndPred P (V.map (subst_term σ) ts) :: Δ))) ->

86 LKID ( Γ ⊢ FIndPred

87 (indcons pr)

88 (V.map (subst_term σ) (indargs pr))

89 :: Δ).

Primjer 16. Produkcijsko pravilo za produkciju PA_prod_E_succ glasi:

Γ ⊢ 𝑂𝑑𝑑(𝑥)[𝜎],∆
(PA_prod_E_succProd)

Γ ⊢ 𝐸𝑣𝑒𝑛(𝑠(𝑥))[𝜎],∆

gdje je 𝑥 proizvoljna varijabla, a 𝜎 proizvoljna supstitucija. Supstitucijom 𝜎 kažemo da

u produkcijama imena varijabli nisu bitna, već samo njihov redoslijed.

3.6. Dokaz i teorem

Prikazali smo sva pravila izvoda sustava 𝐿𝐾𝐼𝐷 te konačno možemo definirati dokaz. Za

proizvoljnu sekventu 𝑠 ≔ (Γ ⊢ ∆), njen dokaz je stanovnik tipa LKID s. Za formulu

𝜑 kažemo da je dokaziva u sustavu 𝐿𝐾𝐼𝐷 te pišemo ⊢ 𝜑 ako postoji dokaz sekvente

[] ⊢ [𝜑].

1 Definition provable (proofsystem : sequent -> Prop) (φ : formula Σ) :=

2 proofsystem ([] ⊢ [φ]).

Neformalni pojam konačnog stabla formaliziran je induktivnim tipom LKID, dok njegovi

konstruktori formaliziraju odgovarajuća pravila izvoda.

Primjer 17. Za proizvoljnu formulu 𝜑, formula 𝜑 ∨ ‵𝜑 je dokaziva u sustavu 𝐿𝐾𝐼𝐷.

1 Lemma LKID_XM : forall φ, provable LKID (FOr φ (FNeg φ)).

(𝐴𝑥)
𝜑 ⊢ 𝜑

(𝑁𝑒𝑔𝑅)
⊢ ‵𝜑, 𝜑

(𝑃𝑒𝑟𝑚)
⊢ 𝜑,‵𝜑

(𝑂𝑟𝑅)
⊢ 𝜑 ∨ ‵𝜑

Primjer 18. Za sustav 𝐿𝐾𝐼𝐷 vrijedi princip eksplozije.

1 Lemma LKID_EXPLOSION : forall φ Δ, LKID ([FAnd φ (FNeg φ)] ⊢ Δ).

(𝐴𝑥)
𝜑 ⊢ 𝜑,∆

(𝑁𝑒𝑔𝐿)
‵𝜑, 𝜑 ⊢ ∆

(𝑃𝑒𝑟𝑚)
𝜑,‵𝜑 ⊢ ∆

(𝐴𝑛𝑑𝐿)
𝜑 ∧ ‵𝜑 ⊢ ∆

Primjer 19. Formula ∀𝑥, 𝐸𝑣𝑒𝑛(𝑥)→ 𝐸𝑣𝑒𝑛(𝑠(𝑠(𝑥))) je dokaziva u sustavu 𝐿𝐾𝐼𝐷𝑃𝐴.

1 Definition Even_succ_succ_Even : formula Σ__PA :=

2 let x := var_term 0 in

3 FAll

4 (FImp

5 (FIndPred PA_Even [x])

6 (FIndPred PA_Even [TFunc PA_succ

7 [TFunc PA_succ

8 [x]]])).

Neformalni dokaz prikazujemo idućim stablom, a radi preglednosti pišemo 𝐸𝑡 umjesto

𝐸𝑣𝑒𝑛(𝑡) te 𝑂𝑡 umjesto 𝑂𝑑𝑑(𝑡) i ne pišemo zagrade u pozivu funkcijskog simbola 𝑠.

(𝐴𝑥)
𝐸𝑥 ⊢ 𝐸𝑥, 𝐸𝑠𝑠𝑥

(𝙿𝙰_𝚙𝚛𝚘𝚍_𝙾_𝚜𝚞𝚌𝚌𝑃𝑟𝑜𝑑)
𝐸𝑥 ⊢ 𝑂𝑠𝑥, 𝐸𝑠𝑠𝑥

(𝐴𝑥)
𝐸𝑥,𝑂𝑠𝑥 ⊢ 𝑂𝑠𝑥

(𝙿𝙰_𝚙𝚛𝚘𝚍_𝙴_𝚜𝚞𝚌𝚌𝑃𝑟𝑜𝑑)
𝐸𝑥,𝑂𝑠𝑥 ⊢ 𝐸𝑠𝑠𝑥

(𝐶𝑢𝑡)
𝐸𝑥 ⊢ 𝐸𝑠𝑠𝑥 (𝐼𝑚𝑝𝑅)
⊢ 𝐸𝑥 → 𝐸𝑠𝑠𝑥 (𝐴𝑙𝑙𝑅)

⊢ ∀𝑥, 𝐸𝑥 → 𝐸𝑠𝑠𝑥

1 Lemma provable_Even_succ_succ_Even :

2 provable LKID__PA Even_succ_succ_Even.

Primjer 20. Formula ∀𝑥,𝑁𝑎𝑡(𝑥) → 𝐸𝑣𝑒𝑛(𝑥) ∨ 𝑂𝑑𝑑(𝑥) je dokaziva u sustavu 𝐿𝐾𝐼𝐷𝑃𝐴.

Neformalni dokaz prikazujemo idućim stablom koristeći iste pokrate kao u primjeru 19,

te 𝑁𝑡 za 𝑁𝑎𝑡(𝑡).

(1)
⊢ 𝐸𝑜

(𝑊𝑘)
⊢ 𝐸𝑜, 𝐸𝑥 ∨ 𝑂𝑥

(𝑂𝑟𝑅1)⊢ 𝐸𝑜 ∨ 𝑂𝑜, 𝐸𝑥 ∨ 𝑂𝑥

(𝐴𝑥)
𝐸𝑦 ⊢ 𝐸𝑦

(2)
𝐸𝑦 ⊢ 𝑂𝑠𝑦

(𝑂𝑟𝑅2)𝐸𝑦 ⊢ 𝐸𝑠𝑦 ∨ 𝑂𝑠𝑦

(𝐴𝑥)
𝑂𝑦 ⊢ 𝑂𝑦

(3)
𝑂𝑦 ⊢ 𝐸𝑠𝑦

(𝑂𝑟𝑅1)𝑂𝑦 ⊢ 𝐸𝑠𝑦 ∨ 𝑂𝑠𝑦
(𝑂𝑟𝐿)

𝐸𝑦 ∨ 𝑂𝑦 ⊢ 𝐸𝑠𝑦 ∨ 𝑂𝑠𝑦
(𝑊𝑘)

𝐸𝑦 ∨ 𝑂𝑦 ⊢ 𝐸𝑠𝑦 ∨ 𝑂𝑠𝑦, 𝐸𝑥 ∨ 𝑂𝑥
(𝐴𝑥)

𝐸𝑥 ∨ 𝑂𝑥 ⊢ 𝐸𝑥 ∨ 𝑂𝑥
(𝑁𝑎𝑡𝐼𝑛𝑑)

𝑁𝑥 ⊢ 𝐸𝑥 ∨ 𝑂𝑥
(𝐼𝑚𝑝𝑅)

𝑁𝑥 → 𝐸𝑥 ∨ 𝑂𝑥
(𝐴𝑙𝑙𝑅)

⊢ ∀𝑥,𝑁𝑥 → 𝐸𝑥 ∨ 𝑂𝑥

Ovdje je pretpostavka indukcije formula 𝐸𝑥 ∨ 𝑂𝑥. Valja primijetiti da smo u dokazu

koristili (označene brojevima) i produkcijska pravila za produkcije:

(1) PA_prod_E_zero,

(2) PA_prod_O_succ i

(3) PA_prod_E_succ.

1 Lemma provable_every_nat_is_even_or_odd :

2 provable LKID__PA every_nat_is_even_or_odd.

3.7. Adekvatnost

Za dokazivače teorema nužno je da dokazivost formule povlači njenu istinitost — u 1.

poglavlju objasnili smo zašto. Ovo svojstvo dokaznog sustava naziva se adekvatnost, a

ovdje ćemo je prikazati na primjeru sustava 𝐿𝐾𝐼𝐷. Za istinitost ćemo promatrati samo

strukture koje su standardnimodeli, jer u takvima interpretacije induktivnih predikatnih

simbola imaju smisla. Napomenimo da je 𝐿𝐾𝐼𝐷 adekvatan i za nestandardne modele.

Definicija 20. Kažemo da je sekventa Γ ⊢ ∆ istinita i pišemo Γ ⊫ ∆ ako i samo ako za

svaki standardni model𝑀 i njegovu okolinu 𝜌 vrijedi:

(∀𝜑, 𝜑 ∈ Γ→ 𝜌 ⊨ 𝜑)→ ∃𝜓, 𝜓 ∈ ∆ ∧ 𝜌 ⊨ 𝜓,

odnosno istinitost svake formule u Γ povlači istinitost neke formule u ∆, u skladu s intu-

itivnim shvaćanjem sekventi.

1 Definition Sat_sequent (s : sequent) : Prop :=

2 let '(Γ ⊢ Δ) := s in

3 forall (M : structure Σ), standard_model Φ M -> forall (ρ : env M),

4 (forall φ, In φ Γ -> ρ ⊨ φ) -> exists ψ, In ψ Δ /\ ρ ⊨ ψ.

Kažemo da vrijedi lokalna adekvatnost za pravilo izvoda ako istinitost (sekventi) pre-

misa povlači istinitost sekvente konkluzije. Lokalna adekvatnost kaže da se istinitost

čuva kod primjena pravila zaključivanja.

Primjer 21. Za pravilo 𝐴𝑥 vrijedi lokalna adekvatnost, to jest ako u listama formula Γ i

∆ postoji neka zajednička formula 𝜑, onda vrijedi Γ ⊫ ∆. Tada je svjedok istinitosti u ∆

upravo formula 𝜑.

1 Lemma LS_Ax : forall Γ Δ φ, In φ Γ -> In φ Δ -> Γ ⊫ Δ.

Ako za svako pravilo izvoda vrijedi lokalna adekvatnost, indukcijom po strukturi do-

kaza slijedi i (globalna) adekvatnost. Glavni cilj ovog poglavlja bio je dokazati lokalne

adekvatnosti svih pravila sustava 𝐿𝐾𝐼𝐷. Cilj je postignut za sva pravila osim pravila in-

dukcije.

Teorem 1. Za svako pravilo izvoda sustava 𝐿𝐾𝐼𝐷 koje nije indukcijsko vrijedi lokalna

adekvatnost. Posljedično, ako je sekventa dokaziva bez primjene indukcijskih pravila,

onda je ona istinita.

4. Ciklički dokazi

Svaki Coqov dokaz koji smo do sada prikazali je konačan term. Svi formalni dokazi u

sustavu 𝐿𝐾𝐼𝐷 (uključujući primjere 19 i 20) su konačni. Ovakve dokaze jednim ime-

nom nazivamo dokazima konačne dubine jer su sve „grane” njihovih dokaznih stabala

konačne. Međutim, možemo promatrati i dokaze beskonačne dubine kod kojih postoje

beskonačne grane. Već smo u odjeljku 1.2. spomenuli koinduktivne tipove, a dokaz be-

skonačne dubine je tada „beskonačni” dokaz neke koinduktivno definirane tvrdnje. U

ovom ćemo kratkom poglavlju ilustrirati dokaze beskonačne dubine te jednu njihovu

posebnu formu: cikličke dokaze.

4.1. Primjer: lijene liste

Definirajmo za početak lijene liste, koje mogu biti konačne, a mogu biti i beskonačne.

Zbog potencijalne beskonačnosti koristimo koinduktivnu definiciju tipa.

1 CoInductive LList (A : Type) :=

2 | LNil : LList A

3 | LCons : A -> LList A -> LList A.

Kažemo da je lijena lista beskonačna ako je njen rep opet beskonačna lijena lista.

1 CoInductive Infinite {A} : LList A -> Prop :=

2 | Infinite_LCons : forall a l, Infinite l -> Infinite (LCons a l).

Ovdje konstruktor Infinite_LConsmožemo shvatiti kao pravilo izvoda. Pokušajmo do-

kazati da je neka intuitivno beskonačna lista i formalno beskonačna. Definiramo funk-

ciju from analogno istoimenoj funkciji iz odjeljka 1.2.

1 CoFixpoint from (n : nat) : LList nat := LCons n (from (S n)).

Intuitivno je jasno da je (za svaki n) lijena lista from n beskonačna pa želimo dokazati

tvrdnju forall n, Infinite (from n).

Prije glavnog dokaza, pogledajmo konkretan primjer na tvrdnji Infinite (from 0).

Po definiciji funkcije from, dovoljno je dokazati tvrdnju Infinite (LCons 0 (from 1)),

a primjenompravila Infinite_LCons dolazimo do tvrdnje Infinite (from 1). Opet je

po definiciji funkcije from dovoljno dokazati tvrdnju Infinite (LCons 1 (from 2)),

a ponovnom primjenom pravila Infinite_LCons dolazimo do tvrdnje Infinite (from

2). Ilustrirajmo ovaj postupak dokaznim stablom.

⋮
Infinite (from 2) (Infinite_LCons)

Infinite (LCons 1 (from 2))
Infinite (from 1) (Infinite_LCons)

Infinite (LCons 0 (from 1))
Infinite (from 0)

Sada je jasno da ćemo dokazati originalnu tvrdnju tek nakon beskonačno mnogo pri-

mjena pravila Infinite_LCons. Ipak, lakše je dokazati tvrdnju forall n, Infinite

(from n). Nakon introdukcije varijable n, odmatanja definicije funkcije from i primjene

pravila Infinite_LCons dolazimo do tvrdnje Infinite (from (S n)).

*
Infinite (from (S n)) (Infinite_LCons)

Infinite (LCons n (from (S n)))
Infinite (from n)

forall n, Infinite (from n)

Između tvrdnji Infinite (from n) i Infinite (from (S n)) „skinut” je prvi član be-

skonačne liste, čime je napravljen napredak u dokazu. Bez napretka nema smisla ponav-

ljati prethodni postupak beskonačno mnogo puta jer se tvrdnja ne mijenja — u suprot-

nom, mogli bismo svaku tvrdnju 𝜑 dokazati trivijalnim dokazom ∗
𝜑 . Uz napredak u

dokazivanju ima smisla pozivati se na tvrdnju koja se dokazuje te takve dokaze nazivamo

cikličkima. Dokaz tvrdnje forall n, Infinite (from n) jedan je primjer cikličkog

dokaza, gdje je simbolom ∗ označeno cikličko „pozivanje” dokaza. U Coqu, napredak je

osiguran uvjetom produktivnosti spomenutim u odjeljku 1.5.

Pažljiv čitatelj će primijetiti da ista argumentacija vrijedi i za tvrdnju Finite (from

n), gdje je predikat Finite definiran na idući način.

1 Inductive Finite {A} : LList A -> Prop :=

2 | Finite_LNil : Finite LNil

3 | Finite_LCons : forall a l, Finite l -> Finite (LCons a l).

Međutim, beskonačne liste po definiciji nisu konačne. Zašto ipak gornji dokazi ne pro-

laze za predikat Finite? Riječ je o načinu definicije; predikat Infinite definiran je

koinduktivno te njegovi stanovnici nisu dobro utemeljeni, dok je predikat Finite defini-

ran induktivno, a njegovi stanovnici jesu dobro utemeljeni. Štoviše, kada bismo predikat

Finite definirali koinduktivno, on bi postao trivijalan, to jest vrijedio bi za proizvoljnu

lijenu listu.

4.2. Primjer: 𝐶𝐿𝐾𝐼𝐷𝜔

Brotherston je formalizirao ideju cikličkih dokaza u dokaznom sustavu 𝐶𝐿𝐾𝐼𝐷𝜔, pod-

sustavu dokaznog sustava 𝐿𝐾𝐼𝐷𝜔, koji je pak proširenje sustava 𝐿𝐾𝐼𝐷 potencijalno be-

skonačnim dokaznim stablima. Dokazni sustav 𝐶𝐿𝐾𝐼𝐷𝜔 ograničen je u odnosu na sus-

tav 𝐿𝐾𝐼𝐷𝜔 tako što svako (potencijalno beskonačno) dokazno stablo smije imati najviše

konačno mnogo različitih podstabala te svaka beskonačna grana tog stabla mora imati

beskonačno mnogo „napredaka”. Dokazni sustav 𝐶𝐿𝐾𝐼𝐷𝜔 prikazat ćemo neformalno.

Primjer 22. Dokazujemo tvrdnju ∀𝑥,𝑁𝑎𝑡(𝑥)→ 𝐸𝑣𝑒𝑛(𝑥) ∨𝑂𝑑𝑑(𝑥) u sustavu 𝐶𝐿𝐾𝐼𝐷𝜔.

Koristimo pokrate kao u primjeru 20.

(PA_prod_E_zeroProd)
⊢ 𝐸𝑜,𝑂𝑜

𝑁𝑥 ⊢ 𝐸𝑥,𝑂𝑥 (†)
(𝑆𝑢𝑏𝑠𝑡)

𝑁𝑦 ⊢ 𝐸𝑦,𝑂𝑦
(𝑃𝑒𝑟𝑚)

𝑁𝑦 ⊢ 𝑂𝑦, 𝐸𝑦
(PA_prod_O_succProd)

𝑁𝑦 ⊢ 𝑂𝑦,𝑂𝑠𝑦
(PA_prod_E_succProd)

𝑁𝑦 ⊢ 𝐸𝑠𝑦, 𝑂𝑠𝑦
(𝐸𝑞𝐿)

𝑥 = 𝑠𝑦,𝑁𝑦 ⊢ 𝐸𝑥,𝑂𝑥
(𝐶𝑎𝑠𝑒 𝑁)

𝑁𝑥 ⊢ 𝐸𝑥,𝑂𝑥 (†)
(𝑂𝑟𝑅)

𝑁𝑥 ⊢ 𝐸𝑥 ∨ 𝑂𝑥 (𝐼𝑚𝑝𝑅)
⊢ 𝑁𝑥 → 𝐸𝑥 ∨ 𝑂𝑥 (𝐴𝑙𝑙𝑅)

⊢ ∀𝑥,𝑁𝑥 → 𝐸𝑥 ∨ 𝑂𝑥

Iako zapisan u konačnom stablu, ovaj dokaz je beskonačan. Čitajući dokaz od dna do

vrha, primjećujemo nove oznake te nova pravila. Za početak, vidimo da su znakom †

označene gornja i donja sekventa 𝑁𝑥 ⊢ 𝐸𝑥,𝑂𝑥. Gornju sekventu nazivamo pupoljkom

(bud), a donju nazivamo pratiteljem (companion). Može se reći da su pupoljak i nje-

gov pratitelj povezani bridom, tvoreći ciklus u dokaznom stablu — odavde dolazi naziv

ciklički dokaz. U stvari se isječak stabla dokaza između pratitelja i pupoljka ponavlja ad

infinitum pa je dokaz beskonačne dubine. Nadalje, vidimo novo pravilo izvoda𝐶𝑎𝑠𝑒 koje

odgovara rastavu po slučajevima, a koristi se umjesto pravila indukcije. Ako se između

pupoljka i njegovog pratitelja javlja primjena pravila 𝐶𝑎𝑠𝑒, kažemo da je napravljen na-

predak. Konačno, vidimo znak jednakosti uz kojeg je vezano lijevo pravilo za jednakost

𝐸𝑞𝐿. Znak jednakosti je za potrebe sustava 𝐿𝐾𝐼𝐷𝜔 i 𝐶𝐿𝐾𝐼𝐷𝜔 (odnosno pravila 𝐶𝑎𝑠𝑒)

ugrađen u definiciju formule.1

Ciklički dokazi su oni dokazi beskonačne dubine u kojima je između svakog pupoljka

i njegovog pratitelja napravljen napredak, odnosno primijenjeno je pravilo 𝐶𝑎𝑠𝑒. U pri-

mjeru 22, pupoljak i njegov pratitelj javljaju se u istoj grani dokaza, no to nije nužno.

Ovu pojavu možemo vidjeti na dokazu obrata sekvente 𝑁𝑥 ⊢ 𝐸𝑥 ∨ 𝑂𝑥.

Primjer 23. Dokazujemo sekventu 𝐸𝑥 ∨ 𝑂𝑥 ⊢ 𝑁𝑥 u sustavu 𝐶𝐿𝐾𝐼𝐷𝜔.

(1)
⊢ 𝑁𝑜 (𝐸𝑞𝐿)

𝑥 = 𝑜 ⊢ 𝑁𝑥

𝑂𝑥 ⊢ 𝑁𝑥 (†)
(𝑆𝑢𝑏𝑠𝑡)

𝑂𝑦 ⊢ 𝑁𝑦
(2)

𝑂𝑦 ⊢ 𝑁𝑠𝑦
(𝐸𝑞𝐿)

𝑥 = 𝑠𝑦, 𝑂𝑦 ⊢ 𝑁𝑥
(𝐶𝑎𝑠𝑒 𝐸)

𝐸𝑥 ⊢ 𝑁𝑥 (∗)

𝐸𝑥 ⊢ 𝑁𝑥 (∗)
(𝑆𝑢𝑏𝑠𝑡)

𝐸𝑦 ⊢ 𝑁𝑦
(3)

𝐸𝑦 ⊢ 𝑁𝑠𝑦
(𝐸𝑞𝐿)

𝑥 = 𝑠𝑦, 𝐸𝑦 ⊢ 𝑁𝑥
(𝐶𝑎𝑠𝑒 𝑂)

𝑂𝑥 ⊢ 𝑁𝑥 (†)
(𝑂𝑟𝐿)

𝐸𝑥 ∨ 𝑂𝑥 ⊢ 𝑁𝑥

Ovdje je znakom † označen par pupoljka i pratitelja sekvente 𝑂𝑥 ⊢ 𝑁𝑥, a znakom ∗

je označen par pupoljka i pratitelja sekvente 𝐸𝑥 ⊢ 𝑁𝑥. Iako se odgovarajući parovi

pupoljaka i pratitelja naizgled javljaju u različitim granama, uz dva „odmatanja” stabla

dokaza vidimo da to u stvari nije slučaj.

(1)
⊢ 𝑁𝑜

(𝐸𝑞𝐿)
𝑥 = 𝑜 ⊢ 𝑁𝑥

𝐸𝑥 ⊢ 𝑁𝑥 (∗)
(𝑆𝑢𝑏𝑠𝑡)

𝐸𝑦 ⊢ 𝑁𝑦
(2)

𝐸𝑦 ⊢ 𝑁𝑠𝑦
(𝐸𝑞𝐿)

𝑥 = 𝑠𝑦, 𝐸𝑦 ⊢ 𝑁𝑥
(𝐶𝑎𝑠𝑒 𝑂)

𝑂𝑥 ⊢ 𝑁𝑥✚✚❩❩(†) (𝑆𝑢𝑏𝑠𝑡)
𝑂𝑦 ⊢ 𝑁𝑦

(2)
𝑂𝑦 ⊢ 𝑁𝑠𝑦

(𝐸𝑞𝐿)
𝑥 = 𝑠𝑦, 𝑂𝑦 ⊢ 𝑁𝑥

(𝐶𝑎𝑠𝑒 𝐸)
𝐸𝑥 ⊢ 𝑁𝑥 (∗)

(1)
⊢ 𝑁𝑜

(𝐸𝑞𝐿)
𝑥 = 𝑜 ⊢ 𝑁𝑥

𝑂𝑥 ⊢ 𝑁𝑥 (†)
(𝑆𝑢𝑏𝑠𝑡)

𝑂𝑦 ⊢ 𝑁𝑦
(2)

𝑂𝑦 ⊢ 𝑁𝑠𝑦
(𝐸𝑞𝐿)

𝑥 = 𝑠𝑦, 𝑂𝑦 ⊢ 𝑁𝑥
(𝐶𝑎𝑠𝑒 𝐸)

𝐸𝑥 ⊢ 𝑁𝑥✚✚❩❩(∗) (𝑆𝑢𝑏𝑠𝑡)
𝐸𝑦 ⊢ 𝑁𝑦

(2)
𝐸𝑦 ⊢ 𝑁𝑠𝑦

(𝐸𝑞𝐿)
𝑥 = 𝑠𝑦, 𝐸𝑦 ⊢ 𝑁𝑥

(𝐶𝑎𝑠𝑒 𝑂)
𝑂𝑥 ⊢ 𝑁𝑥 (†)

(𝑂𝑟𝐿)
𝐸𝑥 ∨ 𝑂𝑥 ⊢ 𝑁𝑥

Pupoljak sekvente 𝐸𝑥 ⊢ 𝑁𝑥 zamijenjen je isječkom stabla od pratitelja sekvente 𝐸𝑥 ⊢

𝑁𝑥 do pupoljka sekvente 𝑂𝑥 ⊢ 𝑁𝑥, Nadalje, pupoljak sekvente 𝑂𝑥 ⊢ 𝑁𝑥 zamijenjen je

1Jednakost se u matematičkoj logici najčešće tretira kao predikatni simbol signature.

isječkom stabla od pratitelja sekvente 𝑂𝑥 ⊢ 𝑁𝑥 do pupoljka sekvente 𝐸𝑥 ⊢ 𝑁𝑥. Dokaz

ne bi bio moguć bez korištenja produkcijskih pravila za produkcije:

(1) PA_prod_N_zero i

(2) PA_prod_N_succ.

Važno je spomenuti da je svojstvo „biti ispravan dokaz u sustavu 𝐶𝐿𝐾𝐼𝐷𝜔” odlučivo

te je na temelju toga napravljen ciklički dokazivač teorema Cyclist2 [19]. Za naš pri-

mjer proširene Peanove signature s produkcijama za induktivne predikatne simbole𝑁𝑎𝑡,

𝐸𝑣𝑒𝑛 i 𝑂𝑑𝑑, definicije produkcija možemo zapisati u konfiguracijsku datoteka fo.defs

na idući način.

Nat {

true => Nat(0) |

Nat(x) => Nat(s(x))

} ;

Even {

true => Even(0) |

Odd(x) => Even(s(x))

} ;

Odd {

Even(x) => Odd(s(x))

} ;

Zatim možemo pokrenuti dokazivač naredbom

./fo_prove -D fo.defs -p -S "Nat(x) |- Even(x) \/ Odd(x)"

i dobiti izlaz programa

0: Nat_0(x) |- Odd_2(x) \/ Even_1(x) (Nat L.Unf.) [1 <{(0, 1)}/{}>, 2 <{(0, 1)}/{(0, 0)}>]

1: Nat_1(0) |- Odd_2(0) \/ Even_1(0) (Even R.Unf.) [3 <{(1, 1)}/{}>]

3: Nat_1(0) |- T \/ Odd_2(0) (Id)

2: Nat_1(s(w)) /\ Nat_0(w) |- Odd_2(s(w)) \/ Even_1(s(w)) (Even R.Unf.) [4 <{(1, 1), (0, 0)}/{}>]

4: Nat_1(s(w)) /\ Nat_0(w) |- Odd_4(w) \/ Odd_2(s(w)) (Odd R.Unf.) [5 <{(1, 1), (0, 0)}/{}>]

5: Nat_1(s(w)) /\ Nat_0(w) |- Even_5(w) \/ Odd_4(w) (Weaken) [6 <{(0, 0)}/{}>]

6: Nat_0(w) |- Odd_2(w) \/ Even_1(w) (Subst) [7 <{(0, 0)}/{}>]

7: Nat_0(x) |- Odd_2(x) \/ Even_1(x) (Backl) [0] <pre={(0, 0)}>

koji predstavlja dokaz sekvente 𝑁𝑎𝑡(𝑥) ⊢ 𝐸𝑣𝑒𝑛(𝑥) ∨ 𝑂𝑑𝑑(𝑥). Možemo primijetiti da je

ovaj generirani dokaz sličan našem dokazu iz primjera 22.

2http://www.cyclist-prover.org/

Zaključak

Cilj ovog rada bio je ponuditi pregled visoke razine na alat za dokazivanje Coq te ga pri-

mijeniti na formalizaciju logike prvog reda s induktivnim definicijama. U tu svrhu smo u

prvom dijelu rada prikazali osnove Coqa i teorije na kojoj se on temelji, dok smo ostatak

rada posvetili logici. Definirali smo osnovne pojmove poput formule, strukture, stan-

dardnog modela te dokaznog sustava i dokazali smo neka njihova svojstva. Kroz cijeli

rad ilustrirali smo definicije i leme neformalnim i formaliziranimprimjerima. Iako je po-

nekad bilo teže razlučiti bitne dijelove neformalne definicije, ipak se pokazalo da je Coq

prikladan alat za formalizaciju logike te autor smatra da je cilj rada uspješno postignut.

Ovim radom smo tek zagrebali površinu formalizacije logike prvog reda s induktiv-

nim definicijama. Prirodni nastavak rada je dokaz potpunosti sustava 𝐿𝐾𝐼𝐷 pomoću

Henkinovih modela, a nakon toga i formalizacija cikličkog dokaznog sustava 𝐶𝐿𝐾𝐼𝐷𝜔.

Nadalje, ima smisla formalizirati dokazivače teorema za sustave 𝐿𝐾𝐼𝐷 i 𝐶𝐿𝐾𝐼𝐷𝜔. Kako

logika prvog reda nije odlučiva, takvi dokazivači teorema trebali bi uvesti dodatne pret-

postavke (kao što je pretpostavka zatvorenog svijeta u Prologu) ili se usmjeriti prema

interaktivnom dokazivanju. Za našu formalizaciju sustava 𝐿𝐾𝐼𝐷, ima smisla iskoristiti

dodatne Coqove mogućnosti, posebno za automatizaciju dokaza jezikom taktika Ltac te

za prikladniju notaciju terma, formula i dokaza.

Po autorovu mišljenju, računalna formalizacija matematike je odličan način razmiš-

ljanja i razumijevanja, posebno kada se koristi zajedno s tradicionalnim učenjem jer tada

pojam koji se proučava postaje jasan u potpunosti — formalizacija nas tjera da budemo

posebice pažljivi. Nadalje, u današnje vrijeme umjetne inteligencije i automatski gene-

riranog koda, dokazivanje točnosti programa postaje bitnije no što je ikad bilo. Za kraj

ćemo reći da je formalizacija proces koji je često naporan ali uvijek ispunjujuć.

Sažetak

Primjene Coq alata za dokazivanje u matematici i računarstvu

Miho Hren

Coq je interaktivni dokazivač teorema temeljen na teoriji tipova. U ovom radu pri-

mijenili smo alat Coq na zadatak formalizacije logike prvog reda s induktivnim definici-

jama𝐹𝑂𝐿𝐼𝐷. Nadalje, definirali smo standardnemodele i dokazni sustav 𝐿𝐾𝐼𝐷 za logiku

𝐹𝑂𝐿𝐼𝐷 te dokazali neka njihova svojstva. Također smo neformalno prikazali cikličke do-

kaze u kontekstu logike 𝐹𝑂𝐿𝐼𝐷. Sve definicije i leme u radu ilustrirali smo primjerima te

formalizacijom u Coqu.

Ključne riječi: Coq; logika prvog reda; induktivne definicije; dokaz

Abstract

Applications of the Coq Proof Assistant in mathematics and

computer science

Miho Hren

Coq is an interactive theorem prover based on type theory. In this thesis we applied

the Coq Proof Assistant to the task of formalizing first order logic with inductive defi-

nitions 𝐹𝑂𝐿𝐼𝐷. Furthermore, we defined standard models and the 𝐿𝐾𝐼𝐷 proof system

for 𝐹𝑂𝐿𝐼𝐷, and we proved some of their properties. We also informally presented cyclic

proofs in the context of𝐹𝑂𝐿𝐼𝐷. We illustrated all the definitions and lemmas in this thesis

with examples and formalizations in Coq.

Keywords: Coq; first order logic; inductive definitions; proof

